

Specification for the FIRRTL Language:
Version 0.1.3
PRE-RELEASE VERSION - DO NOT
DISTRIBUTE

Patrick S. Li Adam M. Izraelevitz
psli@eecs.berkeley.edu adamiz@eecs.berkeley.edu

Jonathan Bachrach
jrb@eecs.berkeley.edu

January 20, 2016

Contents

1	Introduction	5
1.1	Background	5
1.2	Design Philosophy	6
2	Acknowledgements	7
3	Circuits and Modules	8
3.1	Circuits	8
3.2	Modules	8
3.3	Externally Defined Modules	9
4	Types	9
4.1	Ground Types	9
4.1.1	Integer Types	9
4.1.2	Clock Type	10
4.2	Vector Types	10

4.3	Bundle Types	10
4.4	Passive Types	11
4.5	Type Equivalence	12
4.6	Weak Type Equivalence	12
4.6.1	Oriented Types	12
4.6.2	Conversion to Oriented Types	13
4.6.3	Oriented Type Equivalence	13
5	Statements	13
5.1	Connect Statements	13
5.1.1	The Connection Algorithm	14
5.2	Partial Connect Statements	15
5.2.1	The Partial Connection Algorithm	16
5.3	Statement Groups	16
5.3.1	Last Connect Semantics	17
5.4	The Empty Statement	18
5.5	Wires	18
5.6	Registers	19
5.7	Invalidate Statements	19
5.7.1	The Invalidate Algorithm	20
5.8	Nodes	20
5.9	The Conditional Statement	21
5.9.1	Syntactic Shorthands	21
5.9.2	Nested Declarations	23
5.9.3	Initialization Coverage	24
5.9.4	Scoping	24
5.9.5	Conditional Last Connect Semantics	24
5.10	Memories	26
5.10.1	Read Ports	28
5.10.2	Write Ports	28
5.10.3	Readwrite Ports	29
5.10.4	Read Under Write Behaviour	29
5.11	Instances	29
5.12	Stops	30
5.13	Formatted Prints	30
5.13.1	Format Strings	31

6	Expressions	31
6.1	Unsigned Integers	32
6.2	Signed Integers	32
6.3	Unsigned Bits	32
6.4	Signed Bits	33
6.5	References	34
6.6	Subfields	34
6.7	Subindices	34
6.8	Subaccesses	35
6.9	Multiplexors	36
6.10	Conditionally Valid	36
6.11	Primitive Operation	36
7	Primitive Operations	36
7.1	Add Operation	36
7.2	Subtract Operation	36
7.3	Add Wrap Operation	37
7.4	Subtract Wrap Operation	37
7.5	Multiply Operation	37
7.6	Divide Operation	38
7.7	Modulus Operation	38
7.8	Quotient Operation	38
7.9	Remainder Operation	39
7.10	Comparison Operations	39
7.11	Equality Comparison	40
7.12	Not-Equality Comparison	40
7.13	Equivalence Comparison	40
7.14	Not-Equivalence Comparison	40
7.15	Multiplex	41
7.16	Padding Operation	41
7.17	Reinterpret Bits as UInt	41
7.18	Reinterpret Bits as SInt	42
7.19	Shift Left Operation	42
7.20	Shift Right Operation	42
7.21	Dynamic Shift Left Operation	42
7.22	Dynamic Shift Right Operation	43
7.23	Logical Convert to Signed	43
7.24	Negate	43

7.25 Bitwise Operations	44
7.26 Reduce Bitwise Operations	44
7.27 Concatenation	44
7.28 Bit Extraction Operation	44
7.29 Bit Range Extraction Operation	45
8 Gender	45
9 Namespaces	45
10 User Guide - Misc	45
11 TODO	47
12 FIRRTL Forms	48
12.1 Resolved Form	48
12.2 Lowered Form	48
13 Annotations - IN PROGRESS	49
14 Concrete Syntax	50
15 FIRRTL Language Definition	55
15.1 Abstract Syntax Tree	55
15.2 Notation	56
16 Future Plans	57
17 Questions and Answers	57

1 Introduction

1.1 Background

The ideas for FIRRTL originated from work on Chisel, a hardware description language embedded in Scala used for writing highly-parameterized circuit design generators. Chisel designers manipulate circuit components using Scala functions, encode their interfaces in Scala types, and use Scala's object-orientation features to write their own circuit libraries. This form of meta-programming enables expressive, reliable and type-safe generators that improve RTL design productivity and robustness.

The computer architecture research at U.C. Berkeley relied critically on Chisel to allow small teams of graduate students to design sophisticated RTL circuits. Over a three year period with under twelve graduate students, the architecture team has taped-out over ten different designs.

Internally, the investment in developing and learning Chisel was rewarded in huge gains in productivity. However, Chisel's external rate of adoption was slow for the following reasons:

1. Learning a functional programming language (Scala) is difficult for RTL designers with limited software-engineering experience
2. Confounding the previous point, conceptually separating the Chisel HDL from the host language is difficult for new users
3. The output of Chisel (Verilog) is unreadable and slow to simulate
4. Writing custom circuit transformers requires intimate knowledge about the internals of the Chisel compiler
5. Chisel semantics are ill-defined and thus impossible to target from other languages
6. Error checking is unprincipled due to ill-defined semantics resulting in incomprehensible error messages

As a consequence, Chisel needed to be redesigned from the ground up to standardize its semantics, modularize its compilation process for robustness, and cleanly separate its front-end, intermediate representation, and back-ends. A well defined intermediate representation (IR) allows the system to be targeted by other host programming languages, making it possible for RTL designers to work within a language they are already comfortable with. A clearly defined IR with a concrete syntax also allows inspection of the output of circuit generators/transformers thus making clear the distinction

between the host language and the constructed circuit. A clearly defined semantics allows users without knowledge of the compiler implementation to write circuit transformers. Examples include optimization of circuits for simulation speed, and automatic insertion of signal activity counters. An additional benefit of a well defined IR is the structural invariants that can be enforced before and after each compilation stage, resulting in a more robust compiler and structured mechanism for error checking.

1.2 Design Philosophy

FIRRTL (Flexible Intermediate Representation for RTL) represents the standardized elaborated circuit that the Chisel DSL produces. FIRRTL represents the circuit immediately after Chisel’s elaboration but before any circuit simplification. It is designed to resemble the Chisel DSL after all meta-programming has executed. Thus a user program that makes little use of meta-programming facilities should look almost identical to the generated FIRRTL.

For this reason, FIRRTL has first-class support for high-level constructs such as vector types, bundle types, when statements, partial connects, and modules. These high-level constructs are then gradually removed by a sequence of “lowering” transformations. During each lowering transformation the circuit is rewritten into an equivalent circuit using simpler, lower-level constructs. Eventually the circuit is simplified to its most restricted form, resembling a structured netlist, which allows for easy translation to an output language (e.g. Verilog). This form is given the name “lowered FIRRTL” (LoFIRRTL) and is a strict subset of the full FIRRTL language.

Because the host language is now used solely for its meta-programming facilities, the frontend can be very light-weight, and additional frontends in other languages can target FIRRTL and reuse the majority of the compiler toolchain.

Similar to backends, it is often convenient to write transformers that accept only the restricted LoFIRRTL subset. However, the transformed circuit is allowed to contain any FIRRTL construct, as it can be subsequently lowered again. We intentionally designed LoFIRRTL to be a subset of the full FIRRTL language to provide this feature.

2 Acknowledgements

The FIRRTL language could not have been developed without the help of many of the faculty and students in the ASPIRE lab, and the University of California, Berkeley.

This project originated from discussions with our advisor, Jonathan Bachrach, who indicated the need for a structural redesign of the Chisel system around a well-defined intermediate representation. Patrick Li designed and implemented the first prototype of the FIRRTL language, wrote the initial specification for the language, and presented it to the Chisel group consisting of Scott Beamer, David Biancolin, Christopher Celio, Henry Cook, Palmer Dabbelt, Donggyu Kim, Jack Koenig, Martin Maas, Albert Magyar, Colin Schmidt, Andrew Waterman, Yunsup Lee, Richard Lin, Eric Love, Albert Ou, Stephen Twigg, Jim Lawson, Brian Richards, Krste Asanovic, and John Wawrzynek.

Adam Izraelevitz then led the design and reimplemented FIRRTL, and after many discussions with the Chisel group, refined the design to its present version.

The authors would like to thank the following individuals for their contributions to the FIRRTL project:

1. Andrew Waterman: for his many contributions to the design of FIRRTL's constructs, for his work on Chisel 3.0, and for porting architecture research infrastructure
2. Richard Lin: for improving the Chisel 3.0 code base for release quality
3. Jack Koenig: for implementing the FIRRTL parser in Scala
4. Henry Cook: for porting and cleaning up many aspects of Chisel 3.0, including the testing infrastructure and the parameterization library
5. Stephen Twigg: for his expertise in hardware intermediate representations and for providing many corner cases to consider
6. Palmer Dabbelt, Eric Love, Martin Maas, Christopher Celio, and Scott Beamer: for their feedback on previous drafts of the FIRRTL specification

And finally this project would not have been possible without the continuous feedback and encouragement of our advisor, Jonathan Bachrach, and his leadership on and implementation of Chisel 3.0.

Research is partially funded by DARPA Award Number XXXX, the Center for Future Architectures Research, a member of STARnet, a Semiconductor Research Corporation program sponsored by MARCO and DARPA, and ASPIRE Lab industrial sponsors and affiliates Intel, Google, Nokia, NVIDIA, Oracle, and Samsung. Any opinions, findings, conclusions, or recommendations in this paper are solely those of the authors and do not necessarily reflect the position or the policy of the sponsors.

3 Circuits and Modules

3.1 Circuits

All FIRRTL circuits consist of a list of modules, each representing one hardware block that could be instantiated. The circuit must specify the name of the top-level module.

```
circuit MyTop :  
  module MyTop :  
    ...  
    module MyModule :  
      ...
```

All module names in a circuit exist in the same namespace, and thus all modules must have a unique name.

3.2 Modules

Each module has a given name, a list of ports, and a statement representing the circuit connections within the module. A module port is specified by its *direction*, which may be input or output, a name, and the data type for the port. The port names exist in the module identifier namespace, and must be unique.

```
module MyModule :  
  input foo: UInt
```



```
output bar: UInt
bar <= foo
```

Note that a module definition does *not* indicate that the module will be physically present in the final circuit. Refer to the description of the instance statement for details on how to instantiate a module (section ??).

3.3 Externally Defined Modules

Externally defined modules consist of a given name, and a list of ports, whose types and names must match its external definition.

```
module MyExternalModule :
  input foo: UInt
  output bar: UInt
  output baz: SInt
```

4 Types

4.1 Ground Types

All types in FIRRTL are either one of the fundamental ground types or are built up from aggregating other types. There are three ground types in FIRRTL, an unsigned integer type, a signed integer type, and a clock type.

4.1.1 Integer Types

Both unsigned and signed integer types may optionally be given a known positive integer bit width.

```
UInt<10>
SInt<32>
```

Alternatively, if the bit width is omitted, it will be automatically inferred by FIRRTL's width inferencer, as detailed in section ??.

```
UInt
SInt
```

4.1.2 Clock Type

The clock type is used to describe wires and ports meant for carrying clock signals. The usage of components with clock types are restricted. Clock signals cannot be used in most primitive operations, and clock signals can only be connected to components that have been declared with the clock type.

The clock type is specified as follows:

`Clock`

4.2 Vector Types

A vector type is used to express an ordered sequence of elements of a given type. The length of the sequence must be non-negative and known. This is akin to array types in the C programming language.

The following example specifies a ten element vector of 16-bit unsigned integers.

`UInt<16>[10]`

The next example specifies a ten element vector of unsigned integers of omitted but identical bit widths.

`UInt[10]`

Note that any type, including other aggregate types, may be used as the element type in the sequence. The following example specifies a twenty element vector, each of which is a ten element vector of 16-bit unsigned integers.

`UInt<16>[10][20]`

4.3 Bundle Types

A bundle type is used to express a collection of nested and named types. All fields in a bundle type must have a given name, and type.

The following is an example of a possible type for representing a complex number. It has two fields, `real`, and `imag`, both 10-bit signed integers.

`{real:SInt<10>, imag:SInt<10>}`

Additionally, a field may optionally be declared with a *flipped* orientation.

```
{word:UInt<32>, valid:UInt<1>, flip ready:UInt<1>}
```

In a connection between elements declared with the same bundle type, the data carried by the flipped fields flow in the opposite direction as the data carried by the non-flipped fields.

As an example, consider a module output port declared with the following type:

```
output mysignal: {word:UInt<32>, valid:UInt<1>, flip ready:UInt<1>}
```

In a connection to the `mysignal` port, the data carried by the `word` and `valid` subfields will flow out of the module, while data carried by the `ready` subfield will flow into the module. More details about how the bundle field orientation affects connections are explained in section ??.

As in the case of vector types, a bundle field may be declared with any type, including other aggregate types.

```
{real: {word:UInt<32>, valid:UInt<1>, flip ready:UInt<1>}  
  imag: {word:UInt<32>, valid:UInt<1>, flip ready:UInt<1>}}
```

When calculating the final direction of data flow, the orientation of a field is applied recursively to all nested types in the field. As an example, consider the following module port declared with a bundle type containing a nested bundle type.

```
output myport: {a: UInt, flip b: {c: UInt, flip d:UInt}}
```

In a connection to `myport`, the `a` subfield flows out of the module. The `c` subfield contained in the `b` subfield flows into the module, and the `d` subfield contained in the `b` subfield flows out of the module.

Note that within a bundle, all field names must be unique.

4.4 Passive Types

Intuitively, a passive type is defined as a type where all leaf elements have a non-flipped orientation. Thus all ground types are passive types. Vector types are passive if their element type is passive. And bundle types are passive if no fields are flipped and if all field types are passive.

4.5 Type Equivalence

The type equivalence relation is used to determine whether a connection between two elements is legal. See section ?? for further details about connect statements.

An unsigned integer type is always equivalent to another unsigned integer type regardless of bit width, and is not equivalent to any other type. Similarly, a signed integer type is always equivalent to another signed integer type regardless of bit width, and is not equivalent to any other type.

Clock types are only equivalent to clock types, and are not equivalent to any other type.

Two vector types are equivalent if they have the same length, and if their element types are equivalent.

Two bundle types are equivalent when they have the same number of fields, and the *i*'th field from each bundle have matching names and orientations, as well as equivalent types. Consequently, `{a:UInt, b:UInt}` is a different type than `{b:UInt, a:UInt}`. `{a: {flip b:UInt}}` is a different type than `{flip a: {b: UInt}}`.

4.6 Weak Type Equivalence

The weak type equivalence relation is used to determine whether a partial connection between two elements is legal. See section ?? for further details about partial connect statements.

Two types are weakly equivalent if their corresponding oriented types are equivalent.

4.6.1 Oriented Types

The weak type equivalence relation requires first a definition of *oriented types*. Intuitively, an oriented type is a type where all orientation information is collated and coupled with the leaf ground types instead of in bundle fields.

An oriented ground type is an orientation coupled with a ground type. An oriented vector type is an ordered sequence of elements of a given oriented type. The length of an oriented vector type must be known and positive. An oriented bundle type is a collection of fields containing a name and an oriented type, but no orientation.

Applying a flip orientation to an oriented type recursively reverses the orientation of every oriented ground type contained within. Applying a non-flip orientation to an oriented type does nothing.

4.6.2 Conversion to Oriented Types

To convert a ground type to an oriented ground type, attach a non-flip orientation to the ground type.

To convert a vector type to an oriented vector type, convert its element type to an oriented type, and retain its length.

To convert a bundle field to an oriented bundle field, first convert its type to an oriented type. Then apply the field orientation to the oriented type, returning a new oriented type. The new oriented type, and the original field's name combine to create the oriented bundle field. To convert a bundle type to an oriented bundle type, convert each bundle field to an oriented bundle field.

4.6.3 Oriented Type Equivalence

Two oriented ground types are equivalent if their orientations and types are equivalent.

Two oriented vector types are equivalent if their element types are equivalent.

Two oriented bundle types are not equivalent if there exists two fields, one from each oriented bundle type, that have identical names but whose oriented types are not equivalent. Otherwise, the oriented bundle types are equivalent.

As stated earlier, two types are weakly equivalent if their corresponding oriented types are equivalent.

5 Statements

Statements are used to instantiate and connect circuit elements together.

5.1 Connect Statements

The connect statement is used to specify a physically wired connection between two circuit elements.

The following example demonstrates connecting a module's input port to its output port, where port `myinput` is connected to port `myoutput`.

```
module MyModule :  
  input myinput: UInt  
  output myoutput: UInt  
  myoutput <= myinput
```

In order for a connection to be legal the following conditions must hold:

1. The types of the left-hand and right-hand side expressions must be equivalent (see section ?? for details).
2. The bit widths of the two expressions must allow for data to always flow from a smaller to equal size or larger bit width.
3. The gender of the left-hand side expression must be female or bi-gender (see section ?? for an explanation of gender).
4. Either the gender of the right-hand side expression is male or bi-gender, or the right-hand side expression has a passive type.

Connect statements from a narrower ground type element to a wider ground type element will have its value automatically sign-extended to the larger bit width.

5.1.1 The Connection Algorithm

Connect statements between ground types cannot be expanded further.

Connect statements between two vector typed elements recursively connects each subelement in the right-hand side expression to the corresponding subelement in the left-hand side expression.

Connect statements between two bundle typed elements connects the *i*'th field of the right-hand side expression and the *i*'th field of the left-hand side expression. If the *i*'th field is not flipped, then the right-hand side field is connected to the left-hand side field. Conversely, if the *i*'th field is flipped, then the left-hand side field is connected to the right-hand side field.

5.2 Partial Connect Statements

The partial connect statement is also used to specify a physically wired connection between two circuit elements. However, it enforces fewer restrictions on the types and widths of the circuit elements it connects.

In order for a partial connect to be legal the following conditions must hold:

1. The types of the left-hand and right-hand side expressions must be weakly equivalent (see section ?? for details).
2. The gender of the left-hand side expression must be female or bi-gender (see section ?? for an explanation of gender).
3. Either the gender of the right-hand side expression is male or bi-gender, or the right-hand side expression has a passive type.

Partial connect statements from a narrower ground type element to a wider ground type element will have its value automatically sign-extended to the larger bit width. Partial connect statements from a wider ground type element to a narrower ground type element will have its value automatically truncated to fit the smaller bit width.

Intuitively, bundle fields with matching names will be connected appropriately, while bundle fields not present in both types will be ignored. Similarly, vectors with mismatched lengths will be connected up to the shorter length, and the remaining elements are ignored.

The following example demonstrates partially connecting a module's input port to its output port, where port `myinput` is connected to port `myoutput`.

```
module MyModule :
  input myinput: {flip a:UInt, b:UInt[2]}
  output myoutput: {flip a:UInt, b:UInt[3], c:UInt}
  myoutput <- myinput
```

The above example is equivalent to the following:

```
module MyModule :
  input myinput: {flip a:UInt, b:UInt[2]}
  output myoutput: {flip a:UInt, b:UInt[3], c:UInt}
  myinput.a <- myoutput.a
  myoutput.b[0] <- myinput.b[0]
  myoutput.b[1] <- myinput.b[1]
```

For details on the syntax and semantics of the subfield and subindex expressions, see section ??.

5.2.1 The Partial Connection Algorithm

A partial connect statement between two ground type elements connects the right-hand side expression to the left-hand side expression. Conversely, a reverse partial connect statement between two ground type elements connects the left-hand side expression to the right-hand side expression.

A partial (or reverse partial) connect statement between two vector typed elements applies a partial (or reverse partial) connect from the first n subelements in the right-hand side expression to the first n corresponding subelements in the left-hand side expression, where n is the length of the smaller vector.

A partial (or reverse partial) connect statement between two bundle typed elements considers any pair of fields, one from the first bundle type and one from the second, with matching names. If the first field in the pair is not flipped, then we apply a partial (or reverse partial) connect from the right-hand side field to the left-hand side field. However, if the first field is flipped, then we apply a reverse partial (or partial) connect from the right-hand side field to the left-hand side field.

5.3 Statement Groups

An ordered sequence of one or more statements can be grouped into a single statement, called a statement group. The following example demonstrates a statement group composed of three connect statements.

```
module MyModule :
  input a : UInt
  input b : UInt
  output myport1 : UInt
  output myport2 : UInt
  myport1 <= a
  myport1 <= b
  myport2 <= a
```


5.3.1 Last Connect Semantics

Ordering of statements is significant in a statement group. Intuitively, statements execute in order, and the effects of later statements take precedence over earlier ones. In the previous example, in the resultant circuit, port `b` will be connected to `myport1`, and port `a` will be connected to `myport2`.

Note that `connect` and `partial connect` statements have equal priority, and later `connect` or `partial connect` statements always take priority over earlier `connect` or `partial connect` statements. For details on the behavior of conditional statements under last connect semantics, see section ??.

In the case where a `connect` to a circuit element with an aggregate type is followed by a `connect` to a subcomponent of that element, only the connection to the subcomponent is overwritten. Connections to the other subcomponents remain unaffected. In the following example, in the resultant circuit, the `c` subcomponent of port `portx` will be connected to the `c` subcomponent of `myport`, and port `porty` will be connected to the `b` subcomponent of `myport`.

```
module MyModule :
  input portx: {b:UInt, c:UInt}
  input porty: UInt
  output myport: {b:UInt, c:UInt}
  myport <= portx
  myport.b <= porty
```

The above circuit can be rewritten as follows.

```
module MyModule :
  input portx: {b:UInt, c:UInt}
  input porty: UInt
  output myport: {b:UInt, c:UInt}
  myport.b <= porty
  myport.c <= portx.c
```

In the case where a connection to a subcomponent of an aggregate circuit element is followed by a connection to the entire circuit element, the later connection overwrites the earlier connections completely.

```
module MyModule :
  input portx: {b:UInt, c:UInt}
```

```
input porty: UInt
output myport: {b:UInt, c:UInt}
myport.b <= porty
myport <= portx
```

The above circuit can be rewritten as follows.

```
module MyModule :
  input portx: {b:UInt, c:UInt}
  input porty: UInt
  output myport: {b:UInt, c:UInt}
  myport <= portx
```

See section ?? for more details about subcomponent expressions.

5.4 The Empty Statement

The empty statement does nothing and is used simply as a placeholder where a statement is expected. It is specified using the `skip` keyword.

The following example:

```
a <= b
skip
c <= d
```

can be equivalently expressed as:

```
a <= b
c <= d
```

The empty statement is most often used as the alternate branch in a conditional statement, or as a convenient placeholder for removed components during transformational passes. See section ?? for details on the conditional statement.

5.5 Wires

A wire is a named combinational circuit element that can be connected to and from using connect and partial connect statements.

The following example demonstrates instantiating a wire with the given name `mywire` and type `UInt`.

```
wire mywire : UInt
```

5.6 Registers

A register is a named stateful circuit element.

The following example demonstrates instantiating a register with the given name `myreg`, type `SInt`, and is driven by the clock signal `myclock`.

```
wire myclock: Clock
reg myreg: SInt, myclock
```

Optionally, for the purposes of circuit initialization, a register can be declared with a reset signal and value. In the following example, `myreg` is assigned the value `myinit` when the signal `myreset` is high.

```
wire myclock: Clock
wire myreset: UInt<1>
wire myinit: SInt
reg myreg: SInt, myclock, myreset, myinit
```

Note that the clock signal for a register must be of type `clock`, the reset signal must be a single bit `UInt`, and the type of initialization value must match the declared type of the register.

5.7 Invalidate Statements

An invalidate statement is used to indicate that a circuit component contains indeterminate values. It is specified as follows:

```
wire w:UInt
w is invalid
```

Invalidate statements can be applied to any circuit element of any type. However, if the circuit element cannot be connected to, then the statement has no effect on the element. The following example demonstrates the effect of invalidating a variety of circuit elements with aggregate types. See section ?? for details on the algorithm for determining what is invalidated.

```
module MyModule :
  input in:{flip a:UInt, b:UInt}
  output out:{flip a:UInt, b:UInt}
```

```

wire w:{flip a:UInt, b:UInt}
in is invalid
out is invalid
w is invalid

```

is equivalent to the following:

```

module MyModule :
  input in:{flip a:UInt, b:UInt}
  output out:{flip a:UInt, b:UInt}
  wire w:{flip a:UInt, b:UInt}
  in.a is invalid
  out.b is invalid
  w.a is invalid
  w.b is invalid

```

For the purposes of simulation, invalidated elements are initialized to random values, and operations involving indeterminate values produce undefined behaviour. This is useful for early detection of errors in simulation.

5.7.1 The Invalidate Algorithm

Invalidating an element with a ground type indicates that the element's value is indetermined if the element is female or bi-gender. Otherwise, the element is unaffected.

Invalidating an element with a vector type recursively invalidates each subelement in the vector.

Invalidating an element with a bundle type recursively invalidates each subelement in the bundle.

5.8 Nodes

A node is simply a named intermediate value in a circuit. The node must be initialized to a value with a passive type and cannot be connected to.

The following example demonstrates instantiating a node with the given name `mynode` initialized with the output of a multiplexor.

```

wire pred: UInt<1>
wire a: SInt
wire b: SInt
node mynode = mux(pred, a, b)

```

5.9 The Conditional Statement

The conditional statement is used to specify a condition under which connections to previously declared components hold. The condition must have a 1-bit unsigned integer type.

In the following example, the wire `x` is connected to the wire `a` only when the `en` signal is high. Otherwise, the wire `x` is connected to the wire `b`.

```
module mymodule :  
  input a: UInt  
  input b: UInt  
  input en: UInt<1>  
  wire x: UInt  
  when en :  
    x <= a  
  else :  
    x <= b
```

5.9.1 Syntactic Shorthands

The `else` branch of a conditional statement may be omitted, in which case a default `else` branch is supplied consisting of the empty statement.

Thus the following example:

```
module mymodule :  
  input a: UInt  
  input b: UInt  
  input en: UInt<1>  
  wire x: UInt  
  when en :  
    x <= a
```

can be equivalently expressed as:

```
module mymodule :  
  input a: UInt  
  input b: UInt  
  input en: UInt<1>  
  wire x: UInt
```

```
when en :  
  x <= a  
else :  
  skip
```

To aid readability of long chains of conditional statements, the colon following the **else** keyword may be omitted if the **else** branch consists of a single conditional statement.

Thus the following example:

```
module mymodule :  
  input a: UInt  
  input b: UInt  
  input c: UInt  
  input d: UInt  
  input c1: UInt<1>  
  input c2: UInt<1>  
  input c3: UInt<1>  
  
  wire x: UInt  
  when c1 :  
    x <= a  
  else :  
    when c2 :  
      x <= b  
    else :  
      when c3 :  
        x <= c  
      else :  
        x <= d
```

can be equivalently expressed as:

```
module mymodule :  
  input a: UInt  
  input b: UInt  
  input c: UInt  
  input d: UInt
```

```
input c1: UInt<1>
input c2: UInt<1>
input c3: UInt<1>
```

```
wire x: UInt
when c1 :
  x <= a
else when c2 :
  x <= b
else when c3 :
  x <= c
else :
  x <= d
```

5.9.2 Nested Declarations

If a component is declared within a conditional statement, connections to the component are unaffected by the condition. In the following example, register `myreg1` is always connected to `a`, and register `myreg2` is always connected to `b`.

```
module mymodule :
  input a: UInt
  input b: UInt
  input en: UInt<1>
  input clk : Clock
  when en :
    reg myreg1 : UInt, clk
    myreg1 <= a
  else :
    reg myreg2 : UInt, clk
    myreg2 <= b
```

Intuitively, a line can be drawn between a connection (or partial connection) to an element and that element's declaration. All conditional statements that are crossed by the line apply to that connection (or partial connection).

5.9.3 Initialization Coverage

Because of the conditional statement, it is possible to syntactically express circuits containing wires that are only partially connected to an expression. In the following example, the wire *w* is connected to *a* when *en* is asserted high, but it is not specified what *w* is connected to when *en* is low.

```
wire en: UInt<1>
wire w: UInt
wire a: UInt
when en :
  w <= a
```

This is an illegal FIRRTL circuit and an error will be thrown during compilation. All wires, memory ports, instance ports, and module ports that can be connected to must be connected to under all conditions.

5.9.4 Scoping

The conditional statement creates a new *scope* each within its *when* and *else* branches. It is an error to refer to any component declared within a branch after the branch has ended.

5.9.5 Conditional Last Connect Semantics

In the case where a connect to a circuit element is followed by a conditional statement containing a connect to the same element, the connection is overwritten only when the condition holds. Intuitively, a multiplexor is generated such that when the condition is low, the multiplexor returns the old value, and otherwise returns the new value. For details about the multiplexor, see section ??.

The following example:

```
wire a: UInt
wire b: UInt
wire c: UInt<1>
wire w: UInt
w <= a
when c :
  w <= b
```


can be rewritten equivalently using a multiplexor as follows:

```
wire a: UInt
wire b: UInt
wire c: UInt<1>
wire w: UInt
w <= mux(c, b, a)
```

In the case where an invalid statement is followed by a conditional statement containing a connect to the invalidated element, the resulting connection to the element can be expressed using a conditionally valid expression. See section ?? for more details about the conditionally valid expression.

```
wire a: UInt
wire c: UInt<1>
wire w: UInt
w is invalid
when c :
  w <= a
```

can be rewritten equivalently as follows:

```
wire a: UInt
wire c: UInt<1>
wire w: UInt
w <= validif(c, a)
```

The behaviour of conditional connections to circuit elements with aggregate types can be modeled by first expanding each connect into individual connect statements on its ground elements (see section ?? and ?? for the connection and partial connection algorithms) and then applying the conditional last connect semantics.

The following example:

```
wire x: {a:UInt, b:UInt}
wire y: {a:UInt, b:UInt}
wire c: UInt<1>
wire w: {a:UInt, b:UInt}
w <= x
when c :
  w <= y
```

can be rewritten equivalently as follows:

```
wire x: {a:UInt, b:UInt}
wire y: {a:UInt, b:UInt}
wire c: UInt<1>
wire w: {a:UInt, b:UInt}
w.a <= mux(c, y.a, x.a)
w.b <= mux(c, y.b, x.b)
```

Similar to the behavior of aggregate types under last connect semantics (see section ??), the conditional connects to a subcomponent of an aggregate element only generates a multiplexors for the subcomponents that are overwritten.

```
wire x: {a:UInt, b:UInt}
wire y: UInt
wire c: UInt<1>
wire w: {a:UInt, b:UInt}
w <= x
when c :
  w.a <= y
```

```
wire x: {a:UInt, b:UInt}
wire y: UInt
wire c: UInt<1>
wire w: {a:UInt, b:UInt}
w.a <= mux(c, y, x.a)
w.b <= x.b
```

5.10 Memories

A memory is an abstract representation of a hardware memory. It is characterized by the following parameters.

1. A passive type representing the type of each element in the memory.
2. A positive integer representing the number of elements in the memory.
3. A variable number of named ports, each being a read port, a write port, or readwrite port.

4. A non-negative integer indicating the read latency, which is the number of cycles after setting the port's read address before the corresponding element's value can be read from the port's data field.
5. A non-negative integer indicating the write latency, which is the number of cycles after setting the port's write address/data before the corresponding element within the memory holds the new value.
6. A read-under-write flag indicating the behaviour of the memory when a memory location is written to while a read to that location is in progress.

The following example demonstrates instantiating a memory containing 256 complex numbers, each with 16-bit signed integer fields for its real and imaginary components. It has two read ports, `r1` and `r2`, and one write port, `w`. It is combinational read (read latency is zero cycles) and has a write latency of one cycle. Finally, its read-under-write behavior is undefined.

```
mem mymem :
  data-type => {real:SInt<16>, imag:SInt<16>}
  depth => 256
  reader => r1
  reader => r2
  writer => w
  read-latency => 0
  write-latency => 1
  read-under-write => undefined
```

In the example above, the type of `mymem` is:

```
{flip r1: {flip data:{real:SInt<16>, imag:SInt<16>},
  addr:UInt<8>,
  en:UInt<1>,
  clk:Clock}
flip r2: {flip data:{real:SInt<16>, imag:SInt<16>},
  addr:UInt<8>,
  en:UInt<1>,
  clk:Clock}
flip w: {data:{real:SInt<16>, imag:SInt<16>},
  mask:{real:UInt<1>, imag:UInt<1>},
```

```

    addr:UInt<8>,
    en:UInt<1>,
    clk:Clock}}

```

The following sections describe how a memory's field types are calculated and the behavior of each type of memory port.

5.10.1 Read Ports

If a memory is declared with element type `T`, has a size less than or equal to 2^N , then the corresponding type of its read ports is:

```
{flip data:T, addr:UInt<N>, en:UInt<1>, clk:Clock}
```

If the `en` field is high, then the element associated with the address in the `addr` field can be retrieved by reading from the `data` field after the appropriate read latency. If the `en` field is low, then the value in the `data` field, after the appropriate read latency, is undefined. The port is driven by the clock signal in the `clk` field.

5.10.2 Write Ports

If a memory is declared with element type `T`, has a size less than or equal to 2^N , then the corresponding type of its write ports is:

```
{data:T, mask:M, addr:UInt<N>, en:UInt<1>, clk:Clock}
```

where `M` is the mask type calculated from the element type `T`. Intuitively, the mask type mirrors the aggregate structure of the element type except with all ground types replaced with a single bit unsigned integer type. The *non-masked portion* of the data value is defined as the set of data value subcomponents where the corresponding mask subelement is high.

If the `en` field is high, then the non-masked portion of the `data` field value is written, after the appropriate write latency, to the location indicated by the `addr` field. If the `en` field is low, then no value is written after the appropriate write latency. The port is driven by the clock signal in the `clk` field.

5.10.3 Readwrite Ports

Finally, the corresponding type of its readwrite ports is:

```
{rmode:UInt<1>, flip rdata:T, data:T, mask:M,
  addr:UInt<N>, en:UInt<1>, clk:Clock}
```

A readwrite port is a single port that, on a given cycle, can be used either as a write or a read port. If the readwrite port is in read mode (the `rmode` field is high), then the `rdata`, `addr`, `en`, and `clk` fields constitute its read port fields, and should be used accordingly. If the readwrite port is not in read mode (the `rmode` field is low), then the `data`, `mask`, `addr`, `en`, and `clk` fields constitute its write port fields, and should be used accordingly.

5.10.4 Read Under Write Behaviour

The read-under-write flag indicates the resultant value held on the `data` field of a read port if the memory location is written to while a read to that location is in progress. The flag may take on three settings: `old`, `new`, and `undefined`.

If the read-under-write flag is set to `old`, then a read port always returns the value existing in the memory on the same cycle that the read was requested. Intuitively, this is modeled as a combinational read from the memory that is then delayed by the appropriate read latency.

If the read-under-write flag is set to `new`, then a read port always returns the value existing in the memory on the same cycle that the read was made available. Intuitively, this is modeled as a combinational read from the memory after delaying the read address by the appropriate read latency.

If the read-under-write flag is set to `undefined`, then the value held by the read port after the appropriate read latency is undefined.

In all cases, if a memory location is written to by more than one port on the same cycle, the stored value is undefined.

5.11 Instances

FIRRTL modules are instantiated with the instance statement. The following example demonstrates creating an instance named `myinstance` of the `MyModule` module within the top level module `Top`.

```
circuit Top :  
  module MyModule :  
    input a: UInt  
    output b: UInt  
    b <= a  
  module Top :  
    inst myinstance of MyModule
```

The resulting instance has a bundle type, where each field represents a port, and shares its name, in the instantiated module. The fields corresponding to input ports are flipped to indicate their data flows in the opposite direction as the output ports. The `myinstance` instance in the example above has type `{flip a:UInt, b:UInt}`.

Modules have the property that instances can always be *inlined* into the parent module without affecting the semantics of the circuit.

To disallow infinitely recursive hardware, modules cannot contain instances of itself, either directly, or indirectly through instances of other modules it instantiates.

5.12 Stops

The stop statement is used to halt simulations of the circuit. Backends are free to generate hardware to stop a running circuit for the purpose of debugging, but this is not required by the FIRRTL specification.

A stop statement requires a clock signal, a halt condition signal that has a single bit unsigned integer type, and an integer exit code.

```
wire clk:Clock  
wire halt:UInt<1>  
stop(clk, halt, 42)
```

5.13 Formatted Prints

The formatted print statement is used to print a formatted string during simulations of the circuit. Backends are free to generate hardware that relays this information to a hardware test harness, but this is not required by the FIRRTL specification.

A printf statement requires a clock signal, a print condition signal, a format string, and a variable list of argument signals. The conditional signal

must be a single bit unsigned integer type, and the argument signals must each have a ground type.

```
wire clk:Clock
wire condition:UInt<1>
wire a:UInt
wire b:UInt
printf(clk, condition, "a in hex: %x, b in decimal:%d.\n", a, b)
```

On each positive clock edge, when the conditional signal is high, the printf statement prints out the format string where argument placeholders are substituted with the value of the appropriate argument.

5.13.1 Format Strings

Format strings support the following argument placeholders:

- `%b` : Prints the argument in binary
- `%d` : Prints the argument in decimal
- `%x` : Prints the argument in hexadecimal
- `%%` : Prints a single `%` character

Format strings support the following escape characters:

- `\n` : New line
- `\t` : Tab
- `\\` : Back slash
- `\"` : Double quote
- `\'` : Single quote

6 Expressions

FIRRTL expressions are used for creating literal unsigned and signed integers, for referring to a declared circuit component, for statically and dynamically accessing a nested element within a component, and for performing primitive operations.

6.1 Unsigned Integers

A literal unsigned integer can be created given a non-negative integer value and an optional positive bit width. The following example creates a 10-bit unsigned integer representing the number 42.

```
UInt<10>(42)
```

Note that it is an error to supply a bit width that is not large enough to fit the given value. If the bit width is omitted, then FIRRTL's width inferencer to infer the minimum number of bits necessary to fit the given value.

```
UInt(42)
```

6.2 Signed Integers

Similar to unsigned integers, a literal signed integer can be created given an integer value and an optional positive bit width. The following example creates a 10-bit signed integer representing the number -42.

```
SInt<10>(-42)
```

Note that it is an error to supply a bit width that is not large enough to fit the given value using two's complement representation. If the bit width is omitted, then FIRRTL's width inferencer to infer the minimum number of bits necessary to fit the given value.

```
SInt(-42)
```

6.3 Unsigned Bits

A literal unsigned integer can alternatively be created given a string representing its bit representation and an optional bit width.

The following radices are supported:

1. `0b` : For representing binary numbers.
2. `0o` : For representing octal numbers.
3. `0x` : For representing hexadecimal numbers.

If a bit width is not given, the number of bits in the bit representation is directly represented by the string. The following examples create a 8-bit integer representing the number 13.

```
UBits("0b00001101")
UBits("0x0D")
```

If a bit width is given, then the bit representation is truncated to the given bit width. It is an error to supply a bit width that is larger than the number of bits in the bit representation. The following examples create a 7-bit integer representing the number 13.

```
UBits<7>("0b00001101")
UBits<7>("0o015")
UBits<7>("0x0D")
```

6.4 Signed Bits

Similar to unsigned integers, a literal signed integer can alternatively be created given a string representing its bit representation and an optional bit width.

If a bit width is not given, the number of bits in the bit representation is directly represented by the string. The following examples create a 8-bit integer representing the number -13.

```
SBits("0b11110011")
SBits("0xF3")
```

If a bit width is given, then the bit representation is truncated to the given bit width. It is an error to supply a bit width that is larger than the number of bits in the bit representation. The following examples create a 7-bit integer representing the number -13.

```
SBits<7>("0b11110011")
SBits<7>("0o763")
SBits<7>("0xF3")
```

6.5 References

A reference is simply a name that refers to a previously declared circuit component. It may refer to a module port, node, wire, register, instance, or memory.

The following example connects a reference expression `in`, referring to the previously declared port `in`, to the reference expression `out`, referring to the previously declared port `out`.

```
module MyModule :  
  input in : UInt  
  output out : UInt  
  out <= in
```

In the rest of the document, for brevity, the names of components will be used to refer to a reference expression to that component. Thus, the above example will be rewritten as “the port `in` is connected to the port `out`”.

6.6 Subfields

The subfield expression refers to a subelement of an expression with a bundle type.

The following example connects the `in` port to the `a` subelement of the `out` port.

```
module MyModule :  
  input in : UInt  
  output out : {a:UInt, b:UInt}  
  out.a <= in
```

6.7 Subindices

The subindex expression statically refers to a subelement of an expression with a vector type by index. The index must be a non-negative integer and cannot be equal to or exceed the length of the vector it indexes.

The following example connects the `in` port to the fifth subelement of the `out` port.

```

module MyModule :
  input in : UInt
  output out : UInt[10]
  out[4] <= in

```

6.8 Subaccesses

The subaccess expression dynamically refers to a subelement of an expression with a vector type by a calculated index. The index must be an expression with an unsigned integer type.

The following example connects the *n*'th subelement of the *in* port to the *out* port.

```

module MyModule :
  input in: UInt[3]
  input n: UInt<2>
  output out: UInt
  out <= in[n]

```

```

module MyModule :
  input in: UInt[3]
  input n: UInt<2>
  output out: UInt
  when eq(n, UInt(0)) :
    out <= in[0]
  else when eq(n, UInt(1)) :
    out <= in[1]
  else when eq(n, UInt(2)) :
    out <= in[2]
  else :
    out is invalid

```

===== BOOKMARK =====
 TODO connect to a subaccess

6.9 Multiplexors

6.10 Conditionally Valid

6.11 Primitive Operation

$\text{primop}(\text{exp}^*, \text{int}^*)$

There are a number of different primitive operations supported by FIRRTL. Each operation takes some number of expressions, along with some number of integer literals. Section 7 will describe the format and semantics of each operation.

7 Primitive Operations

All primitive operations expression operands must be ground types. In addition, some operations allow all permutations of operand ground types, while others on allow subsets. When well defined, input arguments are allowed to be differing widths, with the semantics of sign-extending the input arguments prior to executing the operation.

7.1 Add Operation

	Input Types	Resultant Type	Resultant Width
add ($op1 : UInt, op2 : UInt$)	$UInt$	$UInt$	$\max(\text{width}(op1), \text{width}(op2)) + 1$
add ($op1 : UInt, op2 : SInt$)	$SInt$	$SInt$	$\max(\text{width}(op1), \text{width}(op2)) + 1$
add ($op1 : SInt, op2 : UInt$)	$SInt$	$SInt$	$\max(\text{width}(op1), \text{width}(op2)) + 1$
add ($op1 : SInt, op2 : SInt$)	$SInt$	$SInt$	$\max(\text{width}(op1), \text{width}(op2)) + 1$

The resultant's value is 1-bit larger than the wider of the two operands and has a signed type if either operand is signed (otherwise is unsigned).

7.2 Subtract Operation

	primop	Resultant Type	Resultant Width
sub ($op1 : UInt, op2 : UInt$)	$SInt$	$SInt$	$\max(\text{width}(op1), \text{width}(op2)) + 1$
sub ($op1 : UInt, op2 : SInt$)	$SInt$	$SInt$	$\max(\text{width}(op1), \text{width}(op2)) + 1$
sub ($op1 : SInt, op2 : UInt$)	$SInt$	$SInt$	$\max(\text{width}(op1), \text{width}(op2)) + 1$
sub ($op1 : SInt, op2 : SInt$)	$SInt$	$SInt$	$\max(\text{width}(op1), \text{width}(op2)) + 1$

The subtraction operation works similarly to the add operation, but always returns a signed integer with a width that is 1-bit wider than the max of the widths of the two operands.

7.3 Add Wrap Operation

	primop	Resultant Type	Resultant Width
addw (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		$\max(\text{width}(\text{op1}), \text{width}(\text{op2}))$
addw (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		$\max(\text{width}(\text{op1}), \text{width}(\text{op2}))$
addw (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>SInt</i>		$\max(\text{width}(\text{op1}), \text{width}(\text{op2}))$
addw (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		$\max(\text{width}(\text{op1}), \text{width}(\text{op2}))$

The add wrap operation works identically to the normal add operation except that the resultant width is the maximum of the width of the two operands, instead of 1 bit greater than the maximum. In the case of overflow, the result silently rolls over.

7.4 Subtract Wrap Operation

	primop	Resultant Type	Resultant Width
subw (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		$\max(\text{width}(\text{op1}), \text{width}(\text{op2}))$
subw (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		$\max(\text{width}(\text{op1}), \text{width}(\text{op2}))$
subw (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>SInt</i>		$\max(\text{width}(\text{op1}), \text{width}(\text{op2}))$
subw (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		$\max(\text{width}(\text{op1}), \text{width}(\text{op2}))$

Similarly to the add wrap operation, the subtract wrap operation works identically to the normal subtract operation except that the resultant width is the maximum of the width of the two operands. In the case of overflow, the result silently rolls over.

7.5 Multiply Operation

	primop	Resultant Type	Resultant Width
mul (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		$\text{width}(\text{op1}) + \text{width}(\text{op2})$
mul (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		$\text{width}(\text{op1}) + \text{width}(\text{op2})$
mul (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>SInt</i>		$\text{width}(\text{op1}) + \text{width}(\text{op2})$
mul (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		$\text{width}(\text{op1}) + \text{width}(\text{op2})$

The resultant value has width equal to the sum of the widths of its two operands.

7.6 Divide Operation

	primop	Resultant Type	Resultant Width
div (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		<i>width</i> (<i>op1</i>)
div (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		<i>width</i> (<i>op1</i>) + 1
div (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>SInt</i>		<i>width</i> (<i>op1</i>)
div (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		<i>width</i> (<i>op1</i>) + 1

The first argument is the dividend, the second argument is the divisor. The resultant width of a divide operation is equal to the width of the dividend, plus one if the divisor is an *SInt*. The resultant value follows the following formula : $\text{div}(a,b) = \text{round-towards-zero}(a/b) + \text{mod}(a,b)$

7.7 Modulus Operation

	primop	Resultant Type	Resultant Width
mod (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		<i>width</i> (<i>op2</i>)
mod (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		<i>width</i> (<i>op2</i>)
mod (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>SInt</i>		<i>width</i> (<i>op2</i>) + 1
mod (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		<i>width</i> (<i>op2</i>)

The first argument is the dividend, the second argument is the divisor. The resultant width of a modulus operation is equal to the width of the divisor, except when the modulus is positive and the result can be negative. The resultant value follows the following formula : $\text{div}(a,b) = \text{round-towards-zero}(a/b) + \text{mod}(a,b)$

7.8 Quotient Operation

	primop	Resultant Type	Resultant Width
quo (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		<i>width</i> (<i>op1</i>) + 1
quo (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		<i>width</i> (<i>op1</i>)
quo (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>SInt</i>		<i>width</i> (<i>op1</i>) + 1
quo (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		<i>width</i> (<i>op1</i>)

The first argument is the dividend, the second argument is the divisor. The resultant width of a quotient operation is equal to the width of the dividend, plus one if the divisor is an *SInt*. The resultant value follows the following formula : $\text{quo}(a,b) = \text{floor}(a/b) + \text{rem}(a,b)$

7.9 Remainder Operation

	primop	Resultant Type	Resultant Width
rem (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		<i>width(op2)</i>
rem (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		<i>width(op2)</i>
rem (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		<i>width(op2)</i> + 1
rem (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		<i>width(op2)</i>

The first argument is the dividend, the second argument is the divisor. The resultant width of a modulus operation is equal to the width of the divisor, except when the divisor is positive and the result can be negative. The resultant value follows the following formula : $\text{quo}(a,b) = \text{floor}(a/b) + \text{rem}(a,b)$

7.10 Comparison Operations

	primop	Resultant Type	Resultant Width
lt (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
lt (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1
lt (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
lt (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1
leq (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
leq (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1
leq (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
leq (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1
gt (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
gt (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1
gt (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
gt (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1
geq (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
geq (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1
geq (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
geq (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1

Each operation accepts any combination of *SInt* or *UInt* input arguments, and always returns a single-bit unsigned integer.

7.11 Equality Comparison

	primop	Resultant Type	Resultant Width
eq (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
eq (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1
eq (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
eq (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1

The equality comparison operator accepts either two unsigned or signed integers and checks whether they are arithmetically equal. The resulting value is a 1-bit unsigned integer.

7.12 Not-Equality Comparison

	primop	Resultant Type	Resultant Width
neq (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
neq (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1
neq (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
neq (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1

The not-equality comparison operator accepts either two unsigned or signed integers and checks whether they are arithmetically not equal. The resulting value is a 1-bit unsigned integer.

7.13 Equivalence Comparison

	primop	Resultant Type	Resultant Width
eqv (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
eqv (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1

The equivalence comparison operator accepts either two unsigned or two signed integers and checks whether they are bitwise equivalent. The resulting value is a 1-bit unsigned integer.

For an arithmetic equals between a signed and unsigned integer, use the equality operator.

7.14 Not-Equivalence Comparison

	primop	Resultant Type	Resultant Width
neqv (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		1
neqv (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>UInt</i>		1

The not-equivalent comparison operator accepts either two unsigned or two signed integers and checks whether they are not bitwise equivalent. The resulting value is a 1-bit unsigned integer.

If an arithmetic not-equals between a signed and unsigned integer is desired, use the not-equals operator.

7.15 Multiplex

	primop	Resultant Type	Resultant Width
mux (<i>condition</i> : <i>UInt</i> , <i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>		$\max(\text{width}(\text{op1}), \text{width}(\text{op2}))$
mux (<i>condition</i> : <i>UInt</i> , <i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>SInt</i>)	<i>SInt</i>		$\max(\text{width}(\text{op1}), \text{width}(\text{op2}))$

The multiplex operation accepts three signals, a 1-bit unsigned integer for the condition expression, followed by either two unsigned integers, or two signed integers. If the condition is high, then the result is equal to the first of the two following operands. If the condition is low, then the result is the second of the two following operands.

The output is of the same width as the max width of the inputs.

7.16 Padding Operation

	primop	Resultant Type	Resultant Width
pad (<i>op</i> : <i>UInt</i> , <i>num</i>)	<i>UInt</i>		<i>num</i>
pad (<i>op</i> : <i>SInt</i> , <i>num</i>)	<i>SInt</i>		<i>num</i>

A pad operation is provided which either zero-extends or sign-extends an expression to a specified width. The given width, *num*, must be equal to or greater than the existing width of the expression.

7.17 Reinterpret Bits as UInt

	primop	Resultant Type	Resultant Width
asUInt (<i>op1</i> : <i>UInt</i>)	<i>UInt</i>		$\text{width}(\text{op1})$
asUInt (<i>op1</i> : <i>SInt</i>)	<i>UInt</i>		$\text{width}(\text{op1})$

Regardless of input type, **primop** returns a *UInt* with the same width as the operand.

7.18 Reinterpret Bits as SInt

	primop	Resultant Type	Resultant Width
	asSInt (<i>op1</i> : <i>UInt</i>)	<i>SInt</i>	<i>width(op1)</i>
	asSInt (<i>op1</i> : <i>SInt</i>)	<i>SInt</i>	<i>width(op1)</i>

Regardless of input type, primop returns a SInt with the same width as the operand.

7.19 Shift Left Operation

	primop	Resultant Type	Resultant Width
	shl (<i>op</i> : <i>UInt</i> , <i>num</i>)	<i>UInt</i>	<i>width(op)</i> + <i>num</i>
	shl (<i>op</i> : <i>SInt</i> , <i>num</i>)	<i>SInt</i>	<i>width(op)</i> + <i>num</i>

The shift left operation accepts either an unsigned or a signed integer, plus a non-negative integer literal specifying the number of bits to shift. The resultant value has the same type as the operand. The output of a shift left operation is equal to the original signal concatenated with *n* zeros at the end, where *n* is the shift amount.

7.20 Shift Right Operation

	primop	Resultant Type	Resultant Width
	shr (<i>op</i> : <i>UInt</i> , <i>num</i>)	<i>UInt</i>	<i>width(op)</i> – <i>num</i>
	shr (<i>op</i> : <i>SInt</i> , <i>num</i>)	<i>SInt</i>	<i>width(op)</i> – <i>num</i>

The shift right operation accepts either an unsigned or a signed integer, plus a non-negative integer literal specifying the number of bits to shift. The resultant value has the same type as the operand. The shift amount must be less than or equal to the width of the operand. The output of a shift right operation is equal to the original signal with the least significant *num* bits truncated, where *num* is the shift amount.

7.21 Dynamic Shift Left Operation

	primop	Resultant Type	Resultant Width
	dshl (<i>op1</i> : <i>UInt</i> , <i>op2</i> : <i>UInt</i>)	<i>UInt</i>	<i>width(op1)</i> + <i>pow</i> (2, <i>width(op2)</i>)
	dshl (<i>op1</i> : <i>SInt</i> , <i>op2</i> : <i>UInt</i>)	<i>SInt</i>	<i>width(op1)</i> + <i>pow</i> (2, <i>width(op2)</i>)

The dynamic shift left operation accepts either an unsigned or a signed integer, plus an unsigned integer dynamically specifying the number of bits to shift. The resultant value has the same type as the operand. The output of a dynamic shift left operation is equal to the original signal concatenated with n zeros at the end, where n is the dynamic shift amount. The output width of a dynamic shift left operation is the width of the original signal plus 2 raised to the width of the dynamic shift amount.

7.22 Dynamic Shift Right Operation

	primop	Resultant Type	Resultant Width
dshr	$(op1 : UInt, op2 : UInt)$	$UInt$	$width(op1)$
dshr	$(op1 : SInt, op2 : UInt)$	$SInt$	$width(op1)$

The shift right operation accepts either an unsigned or a signed integer, plus a non-negative integer literal specifying the number of bits to shift. The resultant value has the same type as the operand. The shift amount must be less than or equal to the width of the operand. The output of a shift right operation is equal to the original signal with the least significant n bits truncated, where n is the dynamic shift amount. The output width of a dynamic shift right operation is the width of the original signal.

7.23 Logical Convert to Signed

	primop	Resultant Type	Resultant Width
cvt	$(op : UInt)$	$SInt$	$width(op) + 1$
cvt	$(op : SInt)$	$SInt$	$width(op)$

The convert operation accepts either an unsigned or a signed integer. The resultant value is always a signed integer. The output of a convert operation will be the same arithmetic value as the input value. The output width is the same as the input width if the input is signed, and increased by one if the input is unsigned.

7.24 Negate

	primop	Resultant Type	Resultant Width
neg	$(op1 : UInt)$	$SInt$	$width(op1) + 1$
neg	$(op1 : SInt)$	$SInt$	$width(op1)$

If the input type is `UInt`, `primop` returns the negative value as an `SInt` with the width of the operand plus one. If the input type is `SInt`, `primop` returns $-1 * \text{input value}$, as an `SInt` with the same width of the operand.

7.25 Bitwise Operations

	primop	Resultant Type	Resultant Width
	not (<i>op1:UInt</i>)	<i>UInt</i>	<i>width(op1)</i>
and (<i>op1:UInt, op2:UInt</i>)		<i>UInt</i>	<i>max(width(op1), width(op2))</i>
or (<i>op1:UInt, op2:UInt</i>)		<i>UInt</i>	<i>max(width(op1), width(op2))</i>
xor (<i>op1:UInt, op2:UInt</i>)		<i>UInt</i>	<i>max(width(op1), width(op2))</i>

The above operations correspond to bitwise not, and, or, and exclusive or respectively. The operands must be unsigned integers, and the resultant width is equal to the width of the wider of the two operands.

7.26 Reduce Bitwise Operations

	primop	Resultant Type	Resultant Width
andr (<i>op:UInt</i>)		<i>UInt</i>	1
orr (<i>op:UInt</i>)		<i>UInt</i>	1
xorr (<i>op:UInt</i>)		<i>UInt</i>	1

The above operations correspond to bitwise not, and, or, and exclusive or respectively, reduced over every bit of a single unsigned integer. The resultant width is always one.

7.27 Concatenation

	primop	Resultant Type	Resultant Width
cat (<i>op1 : UInt, op2 : UInt</i>)		<i>UInt</i>	<i>width(op1) + width(op2)</i>

The concatenation operator accepts two unsigned integers and returns the bitwise concatenation of the two values as an unsigned integer. The resultant width is the sum of the widths of the two operands.

7.28 Bit Extraction Operation

	primop	Resultant Type	Resultant Width
bit (<i>op : UInt, index</i>)		<i>UInt</i>	1

The bit extraction operation accepts an unsigned integer, plus an integer literal specifying the index of the bit to extract. The resultant value is a 1-bit unsigned integer. The index must be non-negative and less than the width of the operand. An index of zero indicates the least significant bit in the operand, and an index of one less than the width the operand indicates the most significant bit in the operand.

7.29 Bit Range Extraction Operation

	primop	Resultant Type	Resultant Width
<code>bits(<i>op</i> : <i>UInt</i>, <i>high</i>, <i>low</i>)</code>		<i>UInt</i>	<i>high</i> − <i>low</i> + 1

The bit range extraction operation accepts either an unsigned integer, plus two integer literals that specify the high (inclusive) and low (inclusive) index of the bit range to extract. The index must be non-negative and less than the width of the operand. Regardless of the type of the operand, the resultant value is a *n*-bit unsigned integer, where *n* = *high* − *low* + 1.

8 Gender

Declared wires are *bidirectional*, which means that they can be used as both an input (by being on the left-hand side of a connect statement), or as an output (by being on the right-hand side of a connect statement).

Like wires, registers are also *bidirectional*, which means that they can be used as both an input (by being on the left-hand side of a connect statement), or as an output (by being on the right-hand side of a connect statement).

9 Namespaces

10 User Guide - Misc

The resulting instance has a bundle type, where the given module's ports are fields and can be accessed using the subfield expression. The orientation of the *output* ports are *default*, and the orientation of the *input* ports are *reverse*. An instance may be directly connected to another element, but it must be on the right-hand side of the connect statement.

The following example illustrates directly connecting an instance to a wire:

```
extmodule Queue :  
  input clk : Clock  
  input in : UInt<16>  
  output out : UInt<16>  
module Top :  
  input clk : Clock  
  inst queue : Queue  
  wire connect : {default out : UInt<16>, reverse in : UInt<16>, reverse clk : Clock}  
  connect <= queue
```

The output ports of an instance may only be connected from, e.g., the right-hand side of a connect statement. Conversely, the input ports of an instance may only be connected to, e.g., the left-hand side of a connect statement.

The following example illustrates a proper use of creating instances with different clock domains:

```

extmodule AsyncQueue :
  input clk1 : Clock
  input clk2 : Clock
  input in : {default data : UInt<16>, reverse ready : UInt<1>}
  output out : {default data : UInt<16>, reverse ready : UInt<1>}
extmodule Source :
  input clk : Clock
  output packet : {default data : UInt<16>, reverse ready : UInt<1>}
extmodule Sink :
  input clk : Clock
  input packet : {default data : UInt<16>, reverse ready : UInt<1>}
module TwoClock :
  input clk1 : Clock
  input clk2 : Clock
  inst src : Source
  src.clk <= clk1
  inst snk : Sink
  snk.clk <= clk2
  inst queue : AsyncQueue
  queue.clk1 <= clk1
  queue.clk2 <= clk2
  queue.in <= src.packet
  snk.packet <= queue.out

```

11 TODO

- Make the syntax for the coding examples consistent.
- Are module names capitalized?
- Is there a space between declared components and the colon before their types.
- etc..
- FIRRTL implementation
- Make register reset/init optional
- Rework readwrite port types
- Add memory read-under-write flag
- Add partial connect algorithm
- Add oriented types to type checker

12 FIRRTL Forms

To simplify the writing of transformation passes, any FIRRTL implementation will provide a *resolving* pass, which resolves all types, widths, and checks the legality of the circuit, and a *lowering* pass, which rewrites any FIRRTL circuit into an equivalent *lowered form*, or LoFIRRTL.

12.1 Resolved Form

The resolved form is guaranteed to be well-formed, meaning all restrictions to a FIRRTL circuit have been checked. In addition, all unknown widths and inferred accessor directions have been resolved.

12.2 Lowered Form

The lowered form, LoFIRRTL, is a structured subset of FIRRTL, making it a minimal representation that is convenient for low-level transforms.

The body of a lowered module consists of a list of declarations, connect statements, and *predicated single connect statements*. A predicated single connect statement is a conditional statement containing a single connect statement and no else branch.

The following circuit is lowered:

```

module MyCounter :
  clock clk : UInt<1>
  input reset : UInt<1>
  input inc : UInt<1>
  output out : UInt<3>
  reg counter : UInt<3>, clk, reset
  when inc : counter <= addw(counter, UInt<1>(1))
  out <= counter

```

The following restrictions also hold for modules in LoFIRRTL.

- **No Nested Expressions** : In the declaration of the structural elements, the only nested expressions allowed are references, and unsigned and signed literals. All other nested expressions must be lifted to a named node, and referred to through a reference.

- **No Composite Types** : No module port or wire may be declared with a bundle or vector type. The lowering pass will recursively expand ports into its constituent elements until all ports are declared with ground types.
- **Single Connect** : Every declared component can only be connected to once within a module. This connect could be a predicated single connect.
- **No Nested Whens** : Other than predicated single connect statements, no other conditional statements are allowed.
- **Inlined Lowered Form** : A further (and optional) pass provided by FIRRTL is the inlining pass, which recursively inlines all instances in the top-level module until the top-level module is the only remaining module in the circuit. Inlined LoFIRRTL is essentially a flat netlist which specifies every component used in a circuit and their input connections.

13 Annotations - IN PROGRESS

Supporting annotations is a critical piece of FIRRTL, yet is a very difficult problem to solve properly. We are in the experimental phase of supporting annotations, and our philosophy is outlined below. It remains to be seen whether our philosophy is correct - if not, we will certainly devise a new strategy.

1. Writing a correct circuit is difficult - avoid silent failures at all costs.
2. If annotations are held in the graph, every pass must properly propagate all possible annotations.
3. A pass incorrectly propagating an annotation cannot be easily detected (silent failure).
4. If annotations are held in an external data structure mapping names to annotations, the structure must be updated after every pass.
5. Incorrectly updating the structure will cause a mismatching of names between circuit components and annotation entries, which is easily detected.

6. Thus, we feel the ability to detect failure outweighs the additional burden on annotation writers.

To implement this philosophy, we encourage passes to either preserve names in the graph, use simple algorithms to transform names, or provide a rename table after a pass. The annotation writer then updates their data structure accordingly.

14 Concrete Syntax

This section describes the text format for FIRRTL that is supported by the provided readers and writers.

General Principles

FIRRTL's text format is human-readable and uses indentation to indicate block structuring. The following characters are allowed in identifiers: upper and lower case letters, digits, as well as the punctuation characters `~!@#$$%^*_+_=?./`. Identifiers cannot begin with a digit.

Comments begin with a semicolon and extend until the end of the line. Commas are treated as whitespace, and may be used by the user for clarity if desired.

Statements are grouped into statement groups using parenthesis, however a colon at the end of a line will automatically assume the next indented region is a statement group. This mechanism is used for indicating block structuring.

The following circuit, module, port and statement examples all exclude the info token `@[filename:line.col]`, which can be optionally included at the beginning of the first line of each elements' concrete syntax.

Circuits and Modules

A circuit is specified the following way.

```
circuit name : (modules ...)
```

Or by taking advantage of indentation structuring:

```
circuit name :  
  modules ...
```

A module is specified the following way.

```
module name : (ports ... stmts ...)
```

The module body consists of a sequence of ports followed immediately by a sequence of statements. If there is more than one statement they are grouped into a statement group by the parser. By using indentation structuring:

```
module name :  
  ports ...  
  stmts ...
```

The following shows an example of a simple module.

```
module mymodule :  
  input a: UInt<1>  
  output b: UInt<1>  
  clock clk: UInt<1>  
  b <= a
```

Types

The unsigned and signed integer types are specified the following way. The following examples demonstrate an unsigned integer with known bit width, signed integer with known bit width, an unsigned integer with unknown bit width, and signed integer with unknown bit width.

```
UInt<42>  
SInt<42>  
UInt<?>  
SInt<?>
```

The bundle type consists of a number of fields surrounded with braces. The following shows an example of a decoupled bundle type. Note that the commas are for clarity only and are not necessary.

```
{default data: UInt<10>,  
  default valid: UInt<1>,  
  reverse ready: UInt<1>}
```

The vector type is specified by immediately postfixing a type with a bracketed integer literal. The following example demonstrates a ten-element vector of 16-bit unsigned integers.

```
UInt<16>[10]
```

Statements

The following examples demonstrate declaring wires, registers, memories, nodes, instances, poisons, and accessors.

```
wire mywire : UInt<10>
reg myreg : UInt<10>, clk, reset
cmem mycombmem : UInt<10>,16
smem myseqmem : UInt<10>,16
inst myinst : MyModule
poison mypoison : UInt<10>
infer accessor myaccessor = e[i],clk
```

The connect statement is specified using the `<=` operator.

```
x <= y
```

The onreset connect statement is specified using the `onreset` keyword and the `<=` operator.

```
onreset x <= y
```

The partial connect statement is specified using the `<-` operator.

```
x <- y
```

The assert statement is specified using the `assert` keyword.

```
assert x
```

The conditional statement is specified with the `when` keyword.

```
when x : x <= y else : x <= z
```

Or by using indentation structuring:

```
when x :  
  x <= y  
else :  
  x <= z
```

If there is no alternative branch specified, the parser will automatically insert an empty statement.

```
when x :  
  x <= y
```

For convenience when expressing nested conditional statements, the colon following the `else` keyword may be elided if the next statement is another conditional statement.

```
when x :  
  x <= y  
else when y :  
  x <= z  
else :  
  x <= w
```

Expressions

The `UInt` and `SInt` constructors create literal integers from a given value and bit width. The following examples demonstrate creating literal integers of both known and unknown bit width.

```
UInt<4>(42)  
SInt<4>(-42)  
UInt<?>(42)  
SInt<?>(-42)
```

References are specified with an identifier.

```
x
```

Subfields are expressed using the dot operator.

```
x.data
```

Subindices are expressed using the `[]` operator.

`x[10]`

Primitive operations are expressed by following the name of the primitive with a list containing the operands.

`add(x, y)`

`add(x, add(x, y))`

`shl(x, 42)`

15 FIRRTL Language Definition

15.1 Abstract Syntax Tree

<i>circuit</i>	=	circuit id : (<i>module</i> [*])	Circuit
<i>module</i>	=	[<i>info</i>] module id : (<i>port</i> [*] <i>stmt</i>)	Module
		[<i>info</i>] extmodule id : (<i>port</i> [*])	External Module
<i>port</i>	=	[<i>info</i>] <i>dir</i> id : <i>type</i>	Port
<i>dir</i>	=	input output	Port Direction
<i>type</i>	=	UInt < <i>width</i> >	Unsigned Integer
		SInt < <i>width</i> >	Signed Integer
		Clock	Clock
		{ <i>field</i> [*] }	Bundle
		<i>type</i> [<i>int</i>]	Vector
<i>field</i>	=	<i>orientation</i> id : <i>type</i>	Bundle Field
<i>orientation</i>	=	default reverse	Orientation
<i>width</i>	=	int ?	Known/Unknown Integer Width
<i>stmt</i>	=	[<i>info</i>] wire id : <i>type</i>	Wire Declaration
		[<i>info</i>] reg id : <i>type</i> , <i>exp</i> , <i>exp</i> , <i>exp</i>	Register Declaration
		[<i>info</i>] mem id : <i>type</i> , int, int, int, (id [*]), (id [*]), (id [*])	Memory Declaration
		[<i>info</i>] inst id : id	Instance Declaration
		[<i>info</i>] poison id : <i>type</i>	Poison Declaration
		[<i>info</i>] node id = <i>exp</i>	Node Declaration
		[<i>info</i>] <i>exp</i> <= <i>exp</i>	Connect
		[<i>info</i>] <i>exp</i> <- <i>exp</i>	Partial Connect
		[<i>info</i>] when <i>exp</i> : <i>stmt</i> else : <i>stmt</i>	Conditional
		[<i>info</i>] stop (<i>exp</i> , <i>exp</i> , int)	Stop Statement
		[<i>info</i>] printf (<i>exp</i> , <i>exp</i> , string, <i>exp</i> [*])	Printf Statement
		[<i>info</i>] skip	Empty Statement
		[<i>info</i>] (<i>stmt</i> [*])	Statement Group
<i>exp</i>	=	[<i>info</i>] UInt < <i>width</i> >(int)	Literal Unsigned Integer
		[<i>info</i>] SInt < <i>width</i> >(int)	Literal Signed Integer
		[<i>info</i>] id	Reference
		[<i>info</i>] <i>exp</i> .id	Subfield
		[<i>info</i>] <i>exp</i> [int]	Subindex
		[<i>info</i>] <i>exp</i> [<i>exp</i>]	Subaccess
		[<i>info</i>] <i>primop</i> (<i>exp</i> [*] , int [*])	Primitive Operation
<i>info</i>	=	filename : line . col	File Location
		noinfo	No File Location

<i>primop</i>	=	add	Unsigned/Signed Add
		sub	Unsigned/Signed Subtract
		addw	Unsigned/Signed Add Wrap
		subw	Unsigned/Signed Subtract Wrap
		mul	Unsigned/Signed Multiply
		div	Unsigned/Signed Divide
		mod	Unsigned/Signed Modulo
		quo	Unsigned/Signed Quotient
		rem	Unsigned/Signed Remainder
		lt	Unsigned/Signed Less Than
		leq	Unsigned/Signed Less or Equal
		gt	Unsigned/Signed Greater Than
		geq	Unsigned/Signed Greater or Equal
		eq	Unsigned/Signed Equal
		neq	Unsigned/Signed Not-Equal
		mux	Unsigned/Signed/Clock Multiplex
		pad	Unsigned/Signed Pad to Length
		asUInt	Unsigned/Signed Reinterpret Bits as UInt
		asSInt	Unsigned/Signed Reinterpret Bits as SInt
		shl	Unsigned/Signed Shift Left
		shr	Unsigned/Signed Shift Right
		dshl	Unsigned/Signed Dynamic Shift Left
		dshr	Unsigned/Signed Dynamic Shift Right
		cvt	Unsigned/Signed to Signed Logical Conversion
		neg	Unsigned/Signed Negate
		not	Unsigned Not
		and	Unsigned And
		or	Unsigned Or
		xor	Unsigned Xor
		andr	Unsigned And Reduce
		orr	Unsigned Or Reduce
		xorr	Unsigned Xor Reduce
		cat	Unsigned Concatenation
		bit	Single Bit Extraction
		bits	Multiple Bit Extraction
		toClock	Interpret Unsigned Bit as Clock
		fromClock	Interpret Clock as Unsigned Bit

15.2 Notation

The above definition specifies the structure of the abstract syntax tree corresponding to a FIRRTL circuit. Nodes in the abstract syntax tree are

italicized. Keywords are shown in **bold**. The special productions `id`, `int`, and `string`, indicates an identifier, an integer literal, and a string respectively. Tokens followed by an asterisk, *e.g.* `field*`, indicates a list formed from repeated occurrences of the token.

Keep in the mind that the above definition is only the *abstract* syntax tree, and is a representation of the in-memory FIRRTL data structure. Readers and writers are provided for converting a FIRRTL data structure into a purely textual representation, which is defined in Section 14.

16 Future Plans

Some choices were made during the design of this specification which were intentionally conservative, so that future versions could lift the restrictions if suitable semantics and implementations are determined. By restricting this version and potentially lifting these restrictions in future versions, all existing FIRRTL circuits will remain valid.

The following design decisions could potentially be changed in future spec revisions:

1. Disallowing zero-width types
2. Always expanding memories into smaller memories (if its type is a non-ground-type)
3. Not including a **ROM** node
4. Custom annotations are not held in FIRRTL nodes
5. Not requiring that all names are unique

17 Questions and Answers

1. Why are there three connect operators? Each is needed for a particular use case - the better question is why did we chose to create multiple connect statements instead of other constructs. Statements, as opposed to expressions, are very restricted in how they nest. Thus, the desired supported behavior (partial connects, full connects, and resets) will never be used in an arbitrary nested expression where the semantics would be unintuitive. In addition, both the implementation and the user only needs to look at the single statement to implement it.

2. Aren't there a lot of idiosyncrasies in FIRRTL? The FIRRTL specification is an ongoing process, and as we push more code through it, it is likely to change. In our opinion, the idiosyncrasies are necessary for a cohesive design (and all languages have idiosyncrasies). It remains an unknown whether there are too many idiosyncrasies for frontend writers. Because the spec is not frozen, we can certainly adapt it if necessary. However, at this point, we just need to push more code through.
3. Why have a separate construct for initializing a register? The problem is initializing a register with a vector/bundle type, where a subset of the fields are initialized. If the initial value is kept with the declaration, we would need a new construct to specify a subset of values of ALL (potentially) nested vector/bundle types. It makes much more sense to separate initialization from the declaration, and use something like a `j=` to initialize the fields/vector sub-components of the register. The next question is why not just have users specify the initial value using their own "when reset :" statement. This doesn't work because of last connect semantics - the user could easily clobber their initialization when statement without knowing. Creating an onreset statement does two things: (1) specifies to the USER exactly what the reset value will be for any sub-component of a register, (2) encapsulates the reset value in a way that is easy for the implementation to special case it (so it doesn't get clobbered).
4. Why do operations allow inputs of differing widths? We tried restricting widths, but it actually complicated width inference and made supporting front-ends with more lax width restrictions very difficult. Because there is perfectly well defined semantics, we opted to allow differing widths. In line with the Linux "funnel" philosophy of being accepting with your inputs and restrictive with your outputs.
5. Why require all names unique? Passes usually need unique names, so there needs to be a renaming pass somewhere. Standardizing how names gets mangled requires a lot of thought, and we didn't feel comfortable putting this into the spec at the moment and potentially regretting it later. For now, names have to be unique, and it is the front-end's responsibility to do this.
6. Why allow declaring components in when statements? We want the important property that a module is just a box of components inside

- for any jumble of components, you can always lace them in the box, and it will preserve the semantics. You need to declare wires inside whens - because generators could run within a when in a front-end. You should always be able to pull them into a module if we want. Now its inconsistent if you can't declare registers in the scope.
- 7. Why not just have LoFIRRTL? LoFIRRTL leaves out general when usage, vector and bundle types, and requires a single connect. For performance backends, we will need to emit arrays and structs. If there is only a lowered circuit, we lose that ability. We cannot simply add vector/bundle types to LoFIRRTL as front-ends cannot easily remove whens without removing the complex types as well. Instead, one will need the expressiveness in FIRRTL to write a performant backend which does not need to operate on LoFIRRTL.
- 8. Why the stop statement have no arguments? Like the enable for write-accessors, the lowering step will preserve the sequence of when statements under which a simulation will stop.
- 9. Why disallow zero-width wires? Very tricky to get the semantics correct. On the todo list.
- 10. Why not require default value for wires? Isn't this a SAT problem? We do the same thing that is done in Java, and is standard programming language practice.
- 11. Why did/didn't you include XXX primop? Up for debate.
- 12. How do you support subword assignment? We decided to not support subword assignment directly, and instead require the user to separate the subword assignment into a vector type. Then, the user uses the subindex expression to assign to an element in the vector.