Specification for the FIRRTL Language: Version 0.1.2 PRE-RELEASE VERSION - DO NOT DISTRIBUTE

Patrick S. Li
psli@eecs.berkeley.edu

Adam M. Izraelevitz
adamiz@eecs.berkeley.edu

Jonathan Bachrach
jrb@eecs.berkeley.edu

June 22, 2015

Contents

1 Introduction - IN PROGRESS

1.1 Background

The ideas for FIRRTL originated from a different UC Berkeley project, Chisel, which embedded a hardware description language in Scala and was used to write highly-parameterized circuit designs.

Paragraph explaining how the concepts of Chisel are great.

1.2 Motivation

Internally, Chisel successfully enabled our research in computer architecture by becoming the backbone of our research infrastructure. Externally, many projects outside of UC Berkeley used Chisel to design their hardware. However, Chisel's external rate of adoption slowed for the following reasons:

1. Learning a functional language (Scala) was a large barrier to entry
2. Conceptually separating the Chisel HDL from the host language was difficult for new users
3. Verilog generation was unreadable and slow
4. Writing transformational passes required insider knowledge of the Chisel compiler
5. Compiler design was unstructured, making error checking difficult and error messages often uncomprehensible
6. Chisel IR semantics were ill-defined and thus impossible to target from other languages

For these reasons, we felt Chisel needed to be redesigned from its ground up to formalize its IR and semantics, modularize its compilation process for robustness, and cleanly separate its frontend (Chisel + Scala), internal representation (FIRRTL), and backends.

1.3 Design Philosophy and Justification

FIRRTL represents the formalized elaborated graph that the Chisel DSL produces, prior to any simplification. By including complicated constructs like vector types, bundle types, and when statements in FIRRTL, the Chisel/Scala frontend can be as light-weight as possible.

Low FIRRTL represents a simplified FIRRTL circuit that lacks any complex features and has additional structural invariants to make it a netlist of a circuit. This form is straightforward to transform into another language (e.g. Verilog) by a light-weight backend.

By containing low FIRRTL within

1. Easy to make light-weight backends
2. Easy to make light-weight front-ends
3. Enables more complicated, but potentially more performant, backends that operate on high firrtl
4. Lowering allows transforms to operate on a limited scope (low firrtl), but produce high firrtl, then lower again if needed.
5. Specification allows additional people to contribute front-ends, backends, and custom transforms.

2 Acknowledgements - IN PROGRESS

The FIRRTL language could not have been developed without the help of many of the faculty and students in the ASPIRE lab, including but not limited to Andrew Waterman, Stephen Twigg, Palmer Dabbelt, Eric Love, Scott Beamer, Chris Celio, Krste Asanovic, and many many others. We'd also like to thank our sponsors XXXX, and the UC Berkeley University.

3 FIRRTL Language Definition

3.1 Abstract Syntax Tree

circuit	=	**circuit** id **:** (*module**)	Circuit
module	=	[*info*] **module** id **:** (*port* stmt*)	Module
	|	[*info*] **exmodule** id **:** (*port**)	External Module
port	=	[*info*] *kind* id **:** *type*	Port
kind	=	**input**|**output**|**clk**	Port Kind
type	=	**UInt**<*width*>	Unsigned Integer
	|	**SInt**<*width*>	Signed Integer
	|	{*field** }	Bundle
	|	*type*[int]	Vector
field	=	*orientation* id **:** *type*	Bundle Field
orientation	=	**default**|**reverse**	Orientation
width	=	int	Known Integer Width
	|	**?**	Unknown Width
stmt	=	[*info*] **wire** id **:** *type*	Wire Declaration
	|	[*info*] **reg** id **:** *type*, id, *exp*	Register Declaration
	|	[*info*] **smem** id **:** *type*, id	Sequential Memory Declaration
	|	[*info*] **cmem** id **:** *type*, id	Combinational Memory Declaration
	|	[*info*] **inst** id **:** id , id*	Instance Declaration
	|	[*info*] **node** id = *exp*	Node Declaration
	|	[*info*] *dir* **accessor** id = *exp*[*exp*]	Accessor Declaration
	|	[*info*] *exp* **:=** *exp*	Connect
	|	[*info*] **onreset** *exp* **:=** *exp*	On-Reset Connect
	|	[*info*] *exp* <> *exp*	Bulk Connect
	|	[*info*] **when** *exp* **:** *stmt* **else :** *stmt*	Conditional
	|	[*info*] *exp*[int **through** int] **:=** *exp*	Sub-Word Assignment
	|	[*info*] **assert** *exp*	Assert Statement
	|	[*info*] **skip**	Empty Statement
	|	[*info*] (*stmt**)	Statement Group
dir	=	**infer**|**read**|**write**|**rdwr**	Accessor Direction
exp	=	[*info*] **UInt**<*width*>(*ints*)	Literal Unsigned Integer
	|	[*info*] **SInt**<*width*>(*ints*)	Literal Signed Integer
	|	[*info*] id	Reference
	|	[*info*] *exp*.id	Subfield
	|	[*info*] *exp*[int]	Subindex
	|	[*info*] *primop*(*exp**, int*)	Primitive Operation
info	=	filename **:** line.col	File Location
	|	**noinfo**	No File Location

primop	=	
	add	Unsigned/Signed Add
|	**sub**	Unsigned/Signed Subtract
|	**addw**	Unsigned/Signed Add Wrap
|	**subw**	Unsigned/Signed Subtract Wrap
|	**mul**	Unsigned/Signed Multiply
|	**div**	Unsigned/Signed Divide
|	**rem**	Unsigned/Signed Remainder
|	**quo**	Unsigned/Signed Quotient
|	**mod**	Unsigned/Signed Modulo
|	**lt**	Unsigned/Signed Less Than
|	**leq**	Unsigned/Signed Less or Equal
|	**gt**	Unsigned/Signed Greater Than
|	**geq**	Unsigned/Signed Greater or Equal
|	**eq**	Unsigned/Signed Equal
|	**neq**	Unsigned/Signed Not-Equal
|	**mux**	Unsigned/Signed Multiplex
|	**pad**	Unsigned/Signed Pad to Length
|	**asUInt**	Unsigned/Signed Reinterpret Bits as UInt
|	**asSInt**	Unsigned/Signed Reinterpret Bits as SInt
|	**shl**	Unsigned/Signed Shift Left
|	**shr**	Unsigned/Signed Shift Right
|	**dshl**	Unsigned/Signed Dynamic Shift Left
|	**dshr**	Unsigned/Signed Dynamic Shift Right
|	**cvt**	Unsigned/Signed to Signed Logical Conversion
|	**neg**	Unsigned/Signed Negate
|	**not**	Unsigned Not
|	**and**	Unsigned And
|	**or**	Unsigned Or
|	**xor**	Unsigned Xor
|	**andr**	Unsigned And Reduce
|	**orr**	Unsigned Or Reduce
|	**xorr**	Unsigned Xor Reduce
|	**cat**	Unsigned Concatenation
|	**bit**	Single Bit Extraction
|	**bits**	Multiple Bit Extraction

3.2 Notation

The above definition specifies the structure of the abstract syntax tree corresponding to a FIRRTL circuit. Nodes in the abstract syntax tree are *italicized*. Keywords are shown in **bold**. The special productions, id and int, indicates an identifier and an integer literal respectively. Tokens followed by

an asterisk, *e.g.* *field**, indicates a list formed from repeated occurences of the token.

Keep in the mind that the above definition is only the *abstract* syntax tree, and is a representation of the in-memory FIRRTL datastructure. Readers and writers are provided for converting a FIRRTL datastructure into a purely textual representation, which is defined in Section 11.

4 Circuits and Modules

circuit	=	**circuit** toplevel-module **:** (modules*)
module	=	**module** name **:** (ports* body)
	|	**exmodule** name **:** (ports*)
port	=	*dir* id **:** *type*
dir	=	**input**|**output**|**clock**

All FIRRTL circuits are comprised of a flat list of modules, each representing one hardware block. Each module has a given name, a list of ports, and a statement representing the circuit connections within the module. Externally defined modules consist of a given name, and a list of ports, whose types must match the types defined in the associated Verilog. Module names exist in their own namespace, and all modules must have a unique name. The name of the top-level module must be specified for a circuit.

A module port is specified by a direction, which may be input or output or clock, a name, and the data type for the port. The port names exist in the identifier namespace for the module, and must be unique. In addition, all references within a module must be unique.

The clock port direction is special in that it cannot be used to connect to any element in the circuit. However, a clock port can be referenced in the **reg**, **cmem**, **smem**, and **inst** declarations, as explained in Section 6.

The following example shows how a module can span two clock domains:

module $TwoClock$:
  clock $clk1$
  clock $clk2$
  input ...

5 Types

5.1 Ground Types

$$\begin{aligned} \textit{type} &= \textbf{UInt}\langle \textit{width} \rangle \\ &\mid \textbf{SInt}\langle \textit{width} \rangle \\ \textit{width} &= \text{int} \\ &\mid \textbf{?} \end{aligned}$$

There are only two ground types in FIRRTL, an unsigned and a signed integer type. Both of these types require a given bitwidth, which may be some known integer width, which must be non-negative and greater than zero, or an unknown width. Unknown widths are a declaration for the width to be computed by the FIRRTL width inferencer, instead of manually given by the programmer. Zero-valued widths are currently not supported, but future versions will likely support them.

5.2 Vector Types

$$\textit{type} = \textit{type}[\text{int}]$$

Vector types in FIRRTL indicate a structure consisting of multiple elements of some given type. This is akin to array types in the C programming language. Note that the number of elements must be known, and non-negative.

As an example, the type **UInt**<16>[10] indicates a ten element vector of 16-bit unsigned integers. The type **UInt**<**?**>[10] indicates a ten element vector of unsigned integers, with unknown but the same bitwidths.

Vector types may be nested ad infinitum. The type **UInt**<16>[10][5] indicates a five element vector *of* ten element vectors of 16-bit unsigned integers.

5.3 Bundle Types

$$\begin{aligned} \textit{type} &= \{\textit{field}*\} \\ \textit{field} &= \textit{orientation}\ \text{name}\ \textbf{:}\ \textit{type} \\ \textit{orientation} &= \textbf{default}|\textbf{reverse} \end{aligned}$$

Bundle types in FIRRTL are composite types formed from an ordered sequence of named, nested types. All fields in a bundle must have a given

direction, name, and type. The following is an example of a possible type for representing a complex number.

{**default** real **:** **SInt**<10>, **default** imag **:** **SInt**<10>}

It has two fields, real, and imag, both 10-bit signed integers. Here is an example of a possible type for a decoupled port.

{**default** data **:** **UInt**<10>,
default valid **:** **UInt**<1>,
reverse ready **:** **UInt**<1>}

It has a data field that is specified to be a 10-bit unsigned integer, a valid signal that must be a 1-bit unsigned integer, and a reversed ready signal that must be a 1-bit unsigned integer.

By convention, we specify the directions within a bundle type with their relative orientation. For this reason, the real and imag fields for the complex number bundle type are both specified to be *default*. Similarly, if a module were to output a value using a decoupled protocol, we would declare the module to have an output port, data, which would contain the value itself, a default field, valid, which would indicate when the value is valid, and accept an *reverse* field, ready, from the receiving component, which would indicate when the component is ready to receive the value.

Note that all field names within a bundle type must be unique.

As in the case of vector types, bundle types may also be nested ad infinitum. I.e., the types of the fields themselves may also be bundle types, which will in turn contain more fields, etc.

6 Statements

FIRRTL circuit components are instantiated and connected together using *statements*.

6.1 Wires

A wire is a named combinational circuit element that can be connected to using the connect statement. A wire with a given name and type can be instantiated with the following statement.

wire name **:** *type*

Declared wires are *bidirectional*, which means that they can be used as both an input (by being on the left-hand side of a connect statement), or as an output (by being on the right-hand side of a connect statement).

6.2 Registers

A register is a named stateful circuit element. A register with a given name, type, clock port name, and reset reference, can be instantiated with the following statement.

reg name **:** *type*, clk, *reset*

Like wires, registers are also *bidirectional*, which means that they can be used as both an input (by being on the left-hand side of a connect statement), or as an output (by being on the right-hand side of a connect statement).

The statement *onreset* is used to specify the initialization value for a register, which is assigned to the register when the declared *reset* signal is asserted.

6.3 Memories

A memory is a stateful circuit element containing multiple elements. Unlike registers, memories can *only* be read from or written to through *accessors*. Memories always have a synchronous write, but can either be declared to be read combinatorially or synchronously. A synchronously read memory with a given name, type and clock port name can be instantiated with the following statement.

smem name **:** *type*, clk

A combinatorially read memory with a given name, type and clock port name can be instantiated with the following statement.

cmem name **:** *type*, clk

Note that, by definition, memories contain multiple elements, and hence *must* be declared with a vector type. Additionally, the type for a memory must be completely specified and cannot contain any unknown widths. It is an error to specify any other type for a memory. However, the internal type to the vector type may be a non-ground type, with the caveat that the internal type, if a bundle type, cannot contain any reverse fields.

A memory cannot be explicitly initialized using a special FIRRTL construct - the circuit itself must contain the proper logic to initialize the memory.

6.4 Nodes

A node is simply a named intermediate value in a circuit, and is akin to a pointer in the C programming language. A node with a given name and value can be instantiated with the following statement.

node name = *exp*

Unlike wires, nodes can only be used in *output* directions. They can be connected from, but not connected to. Consequentially, their expression cannot be a bundle type with any reversed fields.

6.5 Accessors

Accessors are used for either connecting to or from a vector-typed expression, from some *variable* index. An accessor can be instantiated with the following statement.

dir **accessor** name = *exp*[index]

Given a direction, a name, an expression to access, and the index at which to access, the above statement creates an accessor that may be used for connecting to or from the expression. The expression must have a vector type, and the index must be an variable of UInt type.

A read, write, and inferred accessor is conceptually one-way; it must be consistently used to connect to, or to connect from, but not both.

A rdwr is conceptually two-way; it can be used to connect to, to connect from, or both, *but not on the same cycle*. If it is written to and read from on the same cycle, its behavior is undefined.

The following example demonstrates using accessors to read and write to a memory. The accessor, *reader*, acts as a memory read port that reads from the index specified by the wire *i*. The accessor, *writer*, acts as a memory

write port that writes 42 to the index specified by wire j.

wire i : **UInt**<5>
wire j : **UInt**<5>
cmem m : **UInt**<10>[10]
read accessor $reader = m[i]$
write accessor $writer = m[j]$
$writer$:= **UInt**<**?**>(42)
node $temp = reader$

As mentioned previously, the only way to read from or write to a memory is through an accessor. However, accessors are not restricted to accessing memories. They can be used to access *any* vector-valued type.

6.6 Instances

An instance refers to a particular instantiation of a FIRRTL module. An instance with some given name, of a given module, with a list of clock ports can be created using the following statement.

inst name **:** module, clk*

The instance's clock ports are assigned in order of the specified list of enclosing module clock ports. A mismatch in number of clock ports results in an error.

The resulting instance has a bundle type, where the given module's non-clock ports are fields and can be accessed using the subfield expression. The orientation of the *output* ports are *default*, and the orientation of the *input* ports are *reverse*. An instance may be directly connected to another element, but it must be on the right-hand side of the connect statement.

The following example illustrates directly connecting an instance to a wire:

```
exmodule Queue :
  clock clk : UInt<1>
  input in : UInt<16>
  output out : UInt<16>
module Top :
  clock clk : UInt<1>
  inst queue : Queue, clk
  wire connect : {default out : UInt<16>, reverse in: UInt<16>}
  connect := queue
```

The output ports of an instance may only be connected from, e.g. the right-hand side of a connect statement. Conversely, the input ports of an instance may only be connected to, e.g. the left-hand side of a connect statement.

The following example illustrates a proper use of creating instances with different clock domains:

```
exmodule AsyncQueue :
  clock clk1 : UInt<1>
  clock clk2 : UInt<1>
  input in : {default data : UInt<16>, reverse ready : UInt<1>}
  output out : {default data : UInt<16>, reverse ready : UInt<1>}
exmodule Source :
  clock clk : UInt<1>
  output packet : {default data : UInt<16>, reverse ready : UInt<1>}
exmodule Sink :
  clock clk : UInt<1>
  input packet : {default data : UInt<16>, reverse ready : UInt<1>}
module TwoClock :
  clock clk1 : UInt<1>
  clock clk2 : UInt<1>
  inst src : Source, clk1
  inst snk : Sink, clk2
  inst queue : AsynchQueue, clk1, clk2
  queue.in := src.packet
  snk.packet := queue.out
```

There are restrictions upon which modules the user is allowed to instantiate, so as not to create infinitely recursive hardware. We define a module with no instances as a *level 0* module. A module containing only instances of *level 0* modules is a *level 1* module, and a module containing only instances of *level 1* or below modules is a *level 2* module. In general, a *level n* module is only allowed to contain instances of modules of level $n-1$ or below.

6.7 The Connect Statement

The connect statement is used to specify a physical wired connection between one hardware component to another, and is the most important statement in FIRRTL. The following statement is used to connect the output of some component, to the input of another component.

input **:=** output

For a connection to be legal, the types of the two expressions must match exactly, including all field orientations if the elements contain bundle types. The component on the right-hand side must be able to be used as an output, and the component on the left-hand side must be able to be used as an input.

6.8 The Bulk Connect Statement

The bulk connect statement is a connect statement that does not require both expressions to be the same type. During the lowering pass, the bulk connect will expand to some number of connect statements, possibly zero statements. The following statement is used to connect the output of some component, to the input of another component.

input <> output

For a bulk connect between two components of a bundle-type, fields that are of the same type, orientation, and name will be connected. Fields that do not match will not be connected. For a bulk connect between two components of a vector-type, the number of connected elements will be equal to the length of the shorter vector. A bulk connect between two components of the same ground type is equivalent to a normal connect statement. All other combinations of types will not error, but will not generate any connect statements.

6.9 The On-Reset Connect Statement

The on reset connect statement is used to specify the default value for a **reg** element.

onreset r **:=** output

For a connection to be legal, the types of the two expressions must match exactly, including all field orientations if the elements contain bundle types. The component on the right-hand side must be able to be used as an output, and the component on the left-hand side must be a **reg** element. Memories cannot be initialized with this construct.

By default, a **reg** will not have a initialization value and will maintain its current value under the reset signal specified in their declaration. The following example demonstrates declaring a **reg**, and changing its initialization value to forty two.

reg r : **UInt**<10>$(clk,\ reset)$
onreset r := **UInt**<**?**>(42)

6.10 The Sub-Word Connect Statement

The subword connect statement is used to assigned to a range of bits within a ground-typed element. It is specified by two integers that indicate the high and low bounds of the range, inclusively.

exp[hi**through** lo] **:=** output

The subword is always UInt type, so for a connection to be legal, the expression on the right must also be of UInt type. The expression on the right-hand side must be able to be used as an output, and the component on the left-hand side must be able to be used as an input. In addition, the expression on the left-hand side must be a ground type, although this restriction may be relaxed in future spec releases.

6.11 The Conditional Statement

The conditional statement is used to specify a condition that must be asserted under which a list of statements hold. The condition must be a 1-bit unsigned

integer. The following statement states that the *conseq* statements hold only when *condition* is assert high, otherwise the *alt* statements hold instead.

when condition **:** conseq **else :** alt

Notationally, for convenience, we omit the **else** branch if it is an empty statement.

6.11.1 Initialization Coverage

Because of the conditional statement, it is possible for wires to be only partially connected to an expression. In the following example, the wire w is connected to 42 when enable is asserted high, but it is not specified what w is connected to when enable is low. This is an illegal FIRRTL circuit, and will throw a **wire not initialized** error during compilation.

wire w : **UInt**<**?**>
when *enable* :
  w := **UInt**<**?**>(42)

6.11.2 Scoping

The conditional statement creates a new *scope* within its consequent and alternative branches. It is an error to refer to any component declared within a branch after the branch has ended.

Note that there is still only a single identifier namespace in a module. Thus, there cannot be two components with identical names in the same module, *even if* they are in separate scopes. This is to facilitate writing transformational passes, by ensuring that the component name and module name is sufficient to uniquely identify a component.

6.11.3 Connect Semantics - IN PROGRESS

6.12 Statement Groups

Several statements can be grouped into one using the following construct.

(*stmt**)

Ordering is important in a statement group. Later connect statements take precedence over earlier connect statements, and circuit components cannot be referred to before they are instantiated.

6.12.1 Last Connect Semantics

Because of the connect statement, FIRRTL statements are *ordering* dependent. Later connections take precendence over earlier connections. In the following example, the wire w is connected to 42, not 20.

wire w : **UInt**<?>
w := **UInt**<?>(20)
w := **UInt**<?>(42)

By coupling the conditional statement with last connect semantics, many circuits can be expressed in a natural style. In the following example, the wire w is connected to 20 unless the enable expression is asserted high, in which case w is connected to 42.

wire w : **UInt**<?>
w := **UInt**<?>(20)
when *enable* :
  w := **UInt**<?>(42)

6.13 The Assert Statement

The assert statement is used to specify a dynamic assertion on an expression in a circuit.

assert *signal*

The assertion expression must be a one-bit UInt type.

6.14 The Empty Statement

The empty statement is specified using the following.

skip

The empty statement does nothing and is used simply as a placeholder where a statement is expected. It is typically used as the alternative branch in a conditional statement. In addition, it is useful for transformation pass writers.

7 Expressions

FIRRTL expressions are used for creating values corresponding to the ground types, for referring to a declared circuit component, for accessing a nested element within a component, and for performing primitive operations.

7.1 Unsigned Integers

A value of type **UInt** can be directly created using the following expression.

UInt<*width*>(value)

The given value must be non-negative, and the given width, if known, must be large enough to hold the value. If the width is specified as unknown, then FIRRTL infers the minimum possible width necessary to hold the value.

7.2 Signed Integers

A value of type **SInt** can be directly created using the following expression.

SInt<*width*>(value)

The given width, if known, must be large enough to hold the given value in two's complement format. If the width is specified as unknown, then FIRRTL infers the minimum possible width necessary to hold the value.

7.3 References

name

A reference is simply a name that refers to some declared circuit component. A reference may refer to a port, a node, a wire, a register, an instance, a memory, a node, or a structural register.

7.4 Subfields

exp.name

The subfield expression may be used for one of three purposes:

1. To refer to a specific port of an instance, using instance-name.port-name.
2. To refer to a specific field within a bundle-typed expression.

7.5 Subindex

$$exp[\text{index}]$$

The subindex expression is used for referring to a specific element within a vector-valued expression. It is legal to use the subindex expression on any vector-valued expression, except for memories.

7.6 Primitive Operation

$$primop(exp*, \text{int}*)$$

There are a number of different primitive operations supported by FIRRTL. Each operation takes some number of expressions, along with some number of integer literals. Section 8 will describe the format and semantics of each operation.

8 Primitive Operations

All primitive operations expression operands must be ground types. In addition, some allow all permutations of operand ground types, while others on allow subsets. When well defined, input arguments are allowed to be differing widths.

FAQ: Why have generic operations instead of explicit types (e.g. add-uu, where the two inputs are unsigned) FAQ: Why allow operations to allow inputs of differing widths?

8.1 Add Operation

Input Types	Resultant Type	Resultant Width
$\textbf{add}(op1 : UInt, op2 : UInt)$	$UInt$	$max(width(op1), width(op2)) + 1$
$\textbf{add}(op1 : UInt, op2 : SInt)$	$SInt$	$max(width(op1), width(op2)) + 1$
$\textbf{add}(op1 : SInt, op2 : UInt)$	$SInt$	$max(width(op1), width(op2)) + 1$
$\textbf{add}(op1 : SInt, op2 : SInt)$	$SInt$	$max(width(op1), width(op2)) + 1$

The resultant's value is 1-bit larger than the wider of the two operands and has a signed type if either operand is signed (otherwise is unsigned).

8.2 Subtract Operation

primop	Resultant Type	Resultant Width
sub$(op1 : UInt, op2 : UInt)$	$SInt$	$max(width(op1), width(op2)) + 1$
sub$(op1 : UInt, op2 : SInt)$	$SInt$	$max(width(op1), width(op2)) + 1$
sub$(op1 : SInt, op2 : UInt)$	$SInt$	$max(width(op1), width(op2)) + 1$
sub$(op1 : SInt, op2 : SInt)$	$SInt$	$max(width(op1), width(op2)) + 1$

The subtraction operation works similarly to the add operation, but always returns a signed integer with a width that is 1-bit wider than the max of the widths of the two operands.

8.3 Multiply Operation

primop	Resultant Type	Resultant Width
mul$(op1 : UInt, op2 : UInt)$	$UInt$	$width(op1) + width(op2)$
mul$(op1 : UInt, op2 : SInt)$	$SInt$	$width(op1) + width(op2)$
mul$(op1 : SInt, op2 : UInt)$	$SInt$	$width(op1) + width(op2)$
mul$(op1 : SInt, op2 : SInt)$	$SInt$	$width(op1) + width(op2)$

The resultant value has width equal to the sum of the widths of its two operands.

8.4 Divide Operation

primop	Resultant Type	Resultant Width
div$(op1 : UInt, op2 : UInt)$	$UInt$	$width(op1)$
div$(op1 : UInt, op2 : SInt)$	$SInt$	$width(op1) + 1$
div$(op1 : SInt, op2 : UInt)$	$SInt$	$width(op1)$
div$(op1 : SInt, op2 : SInt)$	$SInt$	$width(op1) + 1$

The first argument is the dividend, the second argument is the divisor. The resultant width of a divide operation is equal to the width of the dividend, plus one if the divisor is an SInt. The resultant value follows the following formula : div(a,b) = round-towards-zero(a/b) + mod(a,b)

8.5 Modulus Operation

primop	Resultant Type	Resultant Width
mod($op1 : UInt, op2 : UInt$)	$UInt$	$width(op2)$
mod($op1 : UInt, op2 : SInt$)	$UInt$	$width(op2)$
mod($op1 : SInt, op2 : UInt$)	$SInt$	$width(op2) + 1$
mod($op1 : SInt, op2 : SInt$)	$SInt$	$width(op2)$

The first argument is the dividend, the second argument is the divisor. The resultant width of a modulus operation is equal to the width of the divisor, except when the modulus is positive and the result can be negative. The resultant value follows the following formula : div(a,b) = round-towards-zero(a/b) + mod(a,b)

8.6 Quotient Operation

primop	Resultant Type	Resultant Width
quo($op1 : UInt, op2 : UInt$)	$UInt$	$width(op1) + 1$
quo($op1 : UInt, op2 : SInt$)	$SInt$	$width(op1)$
quo($op1 : SInt, op2 : UInt$)	$SInt$	$width(op1) + 1$
quo($op1 : SInt, op2 : SInt$)	$SInt$	$width(op1)$

The first argument is the dividend, the second argument is the divisor. The resultant width of a quotient operation is equal to the width of the dividend, plus one if the divisor is an SInt. The resultant value follows the following formula : quo(a,b) = floor(a/b) + rem(a,b)

8.7 Remainder Operation

primop	Resultant Type	Resultant Width
rem($op1 : UInt, op2 : UInt$)	$UInt$	$width(op2)$
rem($op1 : UInt, op2 : SInt$)	$SInt$	$width(op2)$
rem($op1 : SInt, op2 : UInt$)	$UInt$	$width(op2) + 1$
rem($op1 : SInt, op2 : SInt$)	$SInt$	$width(op2)$

The first argument is the dividend, the second argument is the divisor. The resultant width of a modulus operation is equal to the width of the divisor, except when the divisor is positive and the result can be negative. The resultant value follows the following formula : quo(a,b) = floor(a/b) + rem(a,b)

8.8 Add Wrap Operation

primop	Resultant Type	Resultant Width
add-wrap$(op1 : UInt, op2 : UInt)$	$UInt$	$max(width(op1), width(op2))$
add-wrap$(op1 : UInt, op2 : SInt)$	$SInt$	$max(width(op1), width(op2))$
add-wrap$(op1 : SInt, op2 : UInt)$	$SInt$	$max(width(op1), width(op2))$
add-wrap$(op1 : SInt, op2 : SInt)$	$SInt$	$max(width(op1), width(op2))$

The add wrap operation works identically to the normal add operation except that the resultant width is the maximum of the width of the two operands, instead of 1 bit greater than the maximum. In the case of overflow, the result silently rolls over.

8.9 Subtract Wrap Operation

primop	Resultant Type	Resultant Width
sub-wrap$(op1 : UInt, op2 : UInt)$	$UInt$	$max(width(op1), width(op2))$
sub-wrap$(op1 : UInt, op2 : SInt)$	$SInt$	$max(width(op1), width(op2))$
sub-wrap$(op1 : SInt, op2 : UInt)$	$SInt$	$max(width(op1), width(op2))$
sub-wrap$(op1 : SInt, op2 : SInt)$	$SInt$	$max(width(op1), width(op2))$

Similarly to the add wrap operation, the subtract wrap operation works identically to the normal subtract operation except that the resultant width is the maximum of the width of the two operands. In the case of overflow, the result silently rolls over.

8.10 Comparison Operations

primop	Resultant Type	Resultant Width
lt$(op1 : UInt, op2 : UInt)$	$UInt$	1
lt$(op1 : UInt, op2 : SInt)$	$UInt$	1
lt$(op1 : SInt, op2 : UInt)$	$UInt$	1
lt$(op1 : SInt, op2 : SInt)$	$UInt$	1
leq$(op1 : UInt, op2 : UInt)$	$UInt$	1
leq$(op1 : UInt, op2 : SInt)$	$UInt$	1
leq$(op1 : SInt, op2 : UInt)$	$UInt$	1
leq$(op1 : SInt, op2 : SInt)$	$UInt$	1
gt$(op1 : UInt, op2 : UInt)$	$UInt$	1
gt$(op1 : UInt, op2 : SInt)$	$UInt$	1
gt$(op1 : SInt, op2 : UInt)$	$UInt$	1
gt$(op1 : SInt, op2 : SInt)$	$UInt$	1
geq$(op1 : UInt, op2 : UInt)$	$UInt$	1
geq$(op1 : UInt, op2 : SInt)$	$UInt$	1
geq$(op1 : SInt, op2 : UInt)$	$UInt$	1
geq$(op1 : SInt, op2 : SInt)$	$UInt$	1

Each operation accept any combination of SInt or UInt input arguements, and always returns a single-bit unsigned integer.

8.11 Equality Comparison

primop	Resultant Type	Resultant Width
eq$(op1 : UInt, op2 : UInt)$	$UInt$	1
eq$(op1 : SInt, op2 : SInt)$	$UInt$	1

The equality comparison operator accepts either two unsigned or two signed integers and checks whether they are bitwise equivalent. The resulting value is a 1-bit unsigned integer.

If an arithmetic equals between a signed and unsigned integer is desired, one must first use convert on the unsigned integer, then use the equal primop.

8.12 Not-Equality Comparison

primop	Resultant Type	Resultant Width
neq$(op1 : UInt, op2 : UInt)$	$UInt$	1
neq$(op1 : SInt, op2 : SInt)$	$UInt$	1

The not-equality comparison operator accepts either two unsigned or two signed integers and checks whether they are not bitwise equivalent. The resulting value is a 1-bit unsigned integer.

If an arithmetic not-equals between a signed and unsigned integer is desired, one must first use convert on the unsigned integer, then use the not-equal primop.

8.13 Multiplex

primop	Resultant Type	Resultant Width
mux(*condition*, *op1*, *op2*)	$UInt$	$width(op1)$
mux(*condition*, *op1*, *op2*)	$SInt$	$width(op1)$

The multiplex operation accepts three signals, a 1-bit unsigned integer for the condition expression, followed by either two unsigned integers, or two signed integers. If the condition is high, then the result is equal to the first of the two following operands. If the condition is low, then the result is the second of the two following operands.

The output is of the same width as the max width of the inputs.

8.14 Padding Operation

primop	Resultant Type	Resultant Width
pad($op : UInt$, num)	$UInt$	num
pad($op : SInt$, num)	$SInt$	num

A pad operation is provided which either zero-extends or sign-extends an expression to a specified width. The given width must be equal to or greater than the existing width of the expression.

8.15 Reinterpret Bits as UInt

primop	Resultant Type	Resultant Width
asUInt($op1 : UInt$)	$UInt$	$width(op1)$
asUInt($op1 : SInt$)	$UInt$	$width(op1)$

Regardless of input type, primop returns a UInt with the same width as the operand.

8.16 Reinterpret Bits as SInt

primop	Resultant Type	Resultant Width
asSInt$(op1 : UInt)$	$SInt$	$width(op1)$
asSInt$(op1 : SInt)$	$SInt$	$width(op1)$

Regardless of input type, primop returns a SInt with the same width as the operand.

8.17 Shift Left Operation

primop	Resultant Type	Resultant Width
shl$(op : UInt, \text{num})$	$UInt$	$width(op) + num$
shl$(op : SInt, \text{num})$	$SInt$	$width(op) + num$

The shift left operation accepts either an unsigned or a signed integer, plus a non-negative integer literal specifying the number of bits to shift. The resultant value has the same type as the operand. The output of a shift left operation is equal to the original signal concatenated with n zeros at the end, where n is the shift amount.

8.18 Shift Right Operation

primop	Resultant Type	Resultant Width
shr$(op : UInt, \text{num})$	$UInt$	$width(op) - num$
shr$(op : SInt, \text{num})$	$SInt$	$width(op) - num$

The shift right operation accepts either an unsigned or a signed integer, plus a non-negative integer literal specifying the number of bits to shift. The resultant value has the same type as the operand. The shift amount must be less than or equal to the width of the operand. The output of a shift right operation is equal to the original signal with the least significant num bits truncated, where num is the shift amount.

8.19 Dynamic Shift Left Operation

primop	Resultant Type	Resultant Width
dshl$(op1 : UInt, op2 : UInt)$	$UInt$	$width(op1) + pow(2, width(op2))$
dshl$(op1 : SInt, op2 : UInt)$	$SInt$	$width(op1) + pow(2, width(op2))$

The dynamic shift left operation accepts either an unsigned or a signed integer, plus an unsigned integer dynamically specifying the number of bits to shift. The resultant value has the same type as the operand. The output of a dynamic shift left operation is equal to the original signal concatenated with n zeros at the end, where n is the dynamic shift amount. The output width of a dynamic shift left operation is the width of the original signal plus 2 raised to the width of the dynamic shift amount.

8.20 Dynamic Shift Right Operation

primop	Resultant Type	Resultant Width
dshr$(op : UInt, op2 : UInt)$	$UInt$	$width(op)$
dshr$(op : SInt, op2 : UInt)$	$SInt$	$width(op)$

The shift right operation accepts either an unsigned or a signed integer, plus a non-negative integer literal specifying the number of bits to shift. The resultant value has the same type as the operand. The shift amount must be less than or equal to the width of the operand. The output of a shift right operation is equal to the original signal with the least significant n bits truncated, where n is the dynamic shift amount. The output width of a dynamic shift right operation is the width of the original signal.

8.21 Convert to Signed

primop	Resultant Type	Resultant Width
convert$(op : UInt)$	$SInt$	$width(op) + 1$
convert$(op : SInt)$	$SInt$	$width(op)$

The convert operation accepts either an unsigned or a signed integer. The resultant value is always a signed integer. The output of a convert operation will be the same arithmetic value as the input value. The output width is the same as the input width if the input is signed, and increased by one if the input is unsigned.

8.22 Negate

primop	Resultant Type	Resultant Width
neg$(op1 : UInt)$	$SInt$	$width(op1) + 1$
neg$(op1 : SInt)$	$SInt$	$width(op1)$

If the input type is UInt, primop returns the negative value as an SInt with the width of the operand plus one. If the input type is SInt, primop returns -1 * input value, as an SInt with the same width of the operand.

8.23 Bitwise Operations

primop	Resultant Type	Resultant Width
not(*op1:UInt*)	$UInt$	$width(op1)$
and(*op1:UInt*, *op2:UInt*)	$UInt$	$max(width(op1), width(op2))$
or(*op1:UInt*, *op2:UInt*)	$UInt$	$max(width(op1), width(op2))$
xor(*op1:UInt*, *op2:UInt*)	$UInt$	$max(width(op1), width(op2))$

The above operations correspond to bitwise not, and, or, and exclusive or respectively. The operands must be unsigned integers, and the resultant width is equal to the width of the wider of the two operands.

8.24 Reduce Bitwise Operations

primop	Resultant Type	Resultant Width
andr(*op:UInt*)	$UInt$	1
orr(*op:UInt*)	$UInt$	1
xorr(*op:UInt*)	$UInt$	1

The above operations correspond to bitwise not, and, or, and exclusive or respectively, reduced over every bit of a single unsigned integer. The resultant width is always one.

8.25 Concatenation

primop	Resultant Type	Resultant Width
cat($op1 : UInt, op2 : UInt$)	$UInt$	$width(op1) + width(op2)$

The concatenation operator accepts two unsigned integers and returns the bitwise concatenation of the two values as an unsigned integer. The resultant width is the sum of the widths of the two operands.

8.26 Bit Extraction Operation

primop	Resultant Type	Resultant Width
bit($op : UInt$, index)	$UInt$	1
bit($op : SInt$, index)	$UInt$	1

The bit extraction operation accepts either an unsigned or a signed integer, plus an integer literal specifying the index of the bit to extract. The resultant value is a 1-bit unsigned integer. The index must be non-negative and less than the width of the operand. An index of zero indicates the least significant bit in the operand, and an index of one less than the width the operand indicates the most significant bit in the operand.

8.27 Bit Range Extraction Operation

primop	Resultant Type	Resultant Width
bits$(op : UInt, \text{high}, \text{low})$	$UInt$	$high - low$
bits$(op : SInt, \text{high}, \text{low})$	$UInt$	$high - low$

The bit range extraction operation accepts either an unsigned or a signed integer, plus two integer literals that specify the high (inclusive) and low (inclusive) index of the bit range to extract. Regardless of the type of the operand, the resultant value is a n-bit unsigned integer, where $n = \text{high} - \text{low} + 1$.

9 FIRRTL Forms

To simplify the writing of transformation passes, FIRRTL provides a *resolving* pass, which resolves all types, widths, and checks the legality of the circuit, and a *lowering* pass, which rewrites any FIRRTL circuit into an equivalent *lowered form*.

9.1 Resolved Circuit

The resolved form is guaranteed to be well-formed, meaning all restrictions to a FIRRTL circuit have been checked. In addition, all unknown widths have been resolved.

9.2 Lowered Circuit

The lowered form has the advantage of not having high-level constructs or composite types, and hence is a minimal representation that is convenient for low-level transforms.

The body of a lowered module consists of a list of declarations, connect statements, and *predicated single connect statements*. A predicated single connect statement is a conditional statement containing a single connect statement and no else branch.

The following circuit is lowered:

```
module MyCounter :
  clock clk : UInt<1>
  input reset : UInt<1>
  input inc : UInt<1>
  output out : UInt<3>
  reg counter : UInt<3>, clk, reset
  when inc : counter := addw(counter, UInt(1))
  out := counter
```

The following restrictions also hold for modules in lowered form.

9.2.1 No Nested Expressions

In the declaration of the structural elements, the only nested expressions allowed are references, and unsigned and signed literals. All other nested expressions must be lifted to a named node, and referred to through a reference.

9.2.2 No Composite Types

No module port or wire may be declared with a bundle or vector type. The lowering pass will recursively expand ports into its constituent elements until all ports are declared with ground types.

9.2.3 Single Connect

Every declared component can only be connected to once within a module. This connect could be a predicated single connect.

9.2.4 No Nested Whens

Other than predicated single connect statements, no other conditional statements are allowed.

9.3 Inlined Lowering Form

A further (and optional) pass provided by FIRRTL is the inlining pass, which recursively inlines all instances in the top-level module until the top-level module is the only remaining module in the circuit. Inlined lowered form is essentially a flat netlist which specifies every component used in a circuit and their input connections.

10 Annotations - IN PROGRESS

1. Writing a correct compiler is difficult.
2. To make it easier, make things as brittle as possible
3. If annotations are kept in the FIRRTL graph, it is unclear how they propagate.
4. If improperly propagated, you either have annotations where they shouldn't be, or lack annotations where they should be.
5. This is impossible to detect, so turns into a silent failure
6. If annotations are used for actual manipulations of circuits later on, this could be the cause of a bug that is exceptionally hard to solve
7. Thus, annotation producer/consumer keeps external datastructure mapping names to annotations
8. Pass writers must do all they can to preserve names - can provide transform for names that annotation users can run on their tables
9. If a name is mangled, the annotation consumer can ERROR. Then, they need to look at the pass to see how their annotations should propagate.

11 Concrete Syntax

This section describes the text format for FIRRTL that is supported by the provided readers and writers.

General Principles

FIRRTL's text format is human-readable and uses indentation to indicate block structuring. The following characters are allowed in identifiers: up-

per and lower case letters, digits, as well as the punctuation characters `~!@#$%^*-_+=?/`. Identifiers cannot begin with a digit.

Comments begin with a semicolon and extend until the end of the line. Commas are treated as whitespace, and may be used by the user for clarity if desired.

Statements are grouped into statement groups using parenthesis, however a colon at the end of a line will automatically surround the next indented region with parenthesis. This mechanism is used for indicating block structuring.

The following circuit, module, port and statement examples all exclude the info token `@[filename:line.col]`, which can be optionally included at the end of the first line of each elements' concrete syntax.

Circuits and Modules

A circuit is specified the following way.

```
circuit name : (modules ...)
```

Or by taking advantage of indentation structuring:

```
circuit name :
   modules ...
```

A module is specified the following way.

```
module name : (ports ... stmts ...)
```

The module body consists of a sequence of ports followed immediately by a sequence of statements. If there is more than one statement they are grouped into a statement group by the parser. By using indentation structuring:

```
module name :
   ports ...
   stmts ...
```

The following shows an example of a simple module.

```
module mymodule :
   input a: UInt<1>
   output b: UInt<1>
   clock clk: UInt<1>
   b := a
```

Types

The unsigned and signed integer types are specified the following way. The following examples demonstrate a unsigned integer with known bitwidth, signed integer with known bitwidth, an unsigned integer with unknown bitwidth, and signed integer with unknown bitwidth.

```
UInt<42>
SInt<42>
UInt<?>
SInt<?>
```

The bundle type consists of a number of fields surrounded with parenthesis. The following shows an example of a decoupled bundle type. Note that the commas are for clarity only and are not necessary.

```
{default data: UInt<10>,
 default valid: UInt<1>,
 reverse ready: UInt<1>}
```

The vector type is specified by immediately postfixing a type with a bracketed integer literal. The following example demonstrates a ten-element vector of 16-bit unsigned integers.

```
UInt<16>[10]
```

Statements

The following examples demonstrate declaring wires, registers, memories, nodes, instances, accessors and bi-accessors.

```
wire mywire : UInt<10>
reg myreg : UInt<10>, clk, reset
cmem mycombmem : UInt<10>[16], clk2
smem myseqmem : UInt<10>[16], clk2
inst myinst : MyModule
infer accessor myaccessor = e[i]
```

The connect statement is specified using the `:=` operator.

```
x := y
```

The on reset connect statement is specified using the onreset keyword and the `:=` operator.

```
onreset x := y
```

The bulk connect statement is specified using the `<>` operator.

```
x <> y
```

The subword connect statement is specified with `[ ]` brackets after the expression on the left of the `:=` operator. Within the brackets is the high bit index, followed by the keyword `through`, followed by the low bit index.

```
x[3 through 0] := y
```

The assert statement is specified using the assert keyword.

```
assert x
```

The conditional statement is specified with the `when` keyword.

```
when x : x := y else : x := z
```

Or by using indentation structuring:

```
when x :
   x := y
else :
   x := z
```

If there is no alternative branch specified, the parser will automatically insert an empty statement.

```
when x :
   x := y
```

For convenience when expressing nested conditional statements, the colon following the `else` keyword may be elided if the next statement is another conditional statement.

```
when x :
   x := y
else when y :
   x := z
else :
   x := w
```

Expressions

The UInt and SInt constructors create literal integers from a given value and bitwidth. The following examples demonstrate creating literal integers of both known and unknown bitwidth.

```
UInt<4>(42)
SInt<4>(-42)
UInt<?>(42)
SInt<?>(-42)
```

References are specified with a identifier.

```
x
```

Subfields are expressed using the dot operator.

```
x.data
```

Subindicies are expressed using the `[]` operator.

```
x[10]
```

Primitive operations are expressed by following the name of the primitive with a list containing the operands.

```
add(x, y)
add(x, add(x, y))
shl(x, 42)
```

12 Future FIRRTL Specification Plans - IN PROGRESS

Some choices were made during the design of this specification which were intentionally conservative, so that future versions could lift the restrictions if suitable semantics and implementations are determined. By restricting this version and potentially lifting them in future versions, all existing FIRRTL circuits will remain valid.

The follow design decisions could potentially be changed in future spec revisions:

1. Disallowing subword assignments to non-ground-typed elements.
2. Disallowing zero-width types
3. Always expanding memories into smaller memories (if the type within the vector type is a non-ground-type)
4. Not including a **printf** node
5. Not including a **ROM** node
6. Custom annotations are not held in FIRRTL nodes
7. Not requiring that all names are unique